

# A Machine-checked Formal Proof of Birkhoff's Variety Theorem in Martin-Löf Type Theory

William DeMeo  

<https://williamdemeo.org>

Jacques Carette  

McMaster University

## 1 Introduction

The Agda Universal Algebra Library ([agda-algebras](#)) is a collection of types and programs (theorems and proofs) formalizing the foundations of universal algebra in dependent type theory using the Agda programming language and proof assistant. The [agda-algebras](#) library now includes a substantial collection of definitions, theorems, and proofs from universal algebra and equational logic and as such provides many examples that exhibit the power of inductive and dependent types for representing and reasoning about general algebraic and relational structures.

The first major milestone of the [agda-algebras](#) project is a new formal proof of *Birkhoff's variety theorem* (also known as the *HSP theorem*), the first version of which was completed in January of 2021. To the best of our knowledge, this was the first time Birkhoff's theorem had been formulated and proved in dependent type theory and verified with a proof assistant.

In this paper, we present a single Agda module called [Demos.HSP](#). This module extracts only those parts of the library needed to prove Birkhoff's variety theorem. In order to meet page limit guidelines, and to reduce strain on the reader, we omit proofs of some routine or technical lemmas that do not provide much insight into the overall development. However, a long version of this paper, which includes all code in the [Demos.HSP](#) module, is available on the arXiv. [reference needed]

In the course of our exposition of the proof of the HSP theorem, we discuss some of the more challenging aspects of formalizing *universal algebra* in type theory and the issues that arise when attempting to constructively prove some of the basic results in this area. We demonstrate that dependent type theory and Agda, despite the demands they place on the user, are accessible to working mathematicians who have sufficient patience and a strong enough desire to constructively codify their work and formally verify the correctness of their results. Perhaps our presentation will be viewed as a sobering glimpse of the painstaking process of doing mathematics in the languages of dependent type theory using the Agda proof assistant. Nonetheless we hope to make a compelling case for investing in these technologies. Indeed, we are excited to share the gratifying rewards that come with some mastery of type theory and interactive theorem proving.

### 1.1 Prior art

There have been a number of efforts to formalize parts of universal algebra in type theory prior to ours, most notably

1. In [2], Capretta formalized the basics of universal algebra in the Calculus of Inductive Constructions using the Coq proof assistant;
2. In [4], Spitters and van der Weegen formalized the basics of universal algebra and some classical algebraic structures, also in the Calculus of Inductive Constructions using the Coq proof assistant and promoting the use of type classes;



This work and the [agda-algebras](#) library by William DeMeo and the [agda-algebras](#) team is licensed under a [Creative Commons Attribution-ShareAlike 4.0 International License](#).

3. In [3] Gunther, et al developed what was (prior to the `agda-algebras` library) the most extensive library of formalized universal algebra to date; like `agda-algebras`, that work is based on dependent type theory, is programmed in Agda, and goes beyond the Noether isomorphism theorems to include some basic equational logic; although the coverage is less extensive than that of `agda-algebras`, Gunther et al do treat *multisorted* algebras, whereas `agda-algebras` is currently limited to single sorted structures.
4. Lynge and Spitters [Lynge:2019] (2019) formalize basic, mutisorted universal algebra, up to the Noether isomorphism theorems, in homotopy type theory; in this setting, the authors can avoid using setoids by postulating a strong extensionality axiom called *univalence*.

Some other projects aimed at formalizing mathematics generally, and algebra in particular, have developed into very extensive libraries that include definitions, theorems, and proofs about algebraic structures, such as groups, rings, modules, etc. However, the goals of these efforts seem to be the formalization of special classical algebraic structures, as opposed to the general theory of (universal) algebras. Moreover, the part of universal algebra and equational logic formalized in the `agda-algebras` library extends beyond the scope of prior efforts.

## 2 Preliminaries

### 2.1 Logical foundations

An Agda program typically begins by setting some language options and by importing types from existing Agda libraries. The language options are specified using the `OPTIONS` pragma which affect control the way Agda behaves by controlling the deduction rules that are available to us and the logical axioms that are assumed when the program is type-checked by Agda to verify its correctness. Every Agda program in the `agda-algebras` library, including the present module (`Demos.HSP`), begins with the following line.

```
{-# OPTIONS -without-K -exact-split -safe #-}
```

We give only very terse descriptions of these options, and refer the reader to the accompanying links for more details.

- *without-K* disables Streicher’s K axiom; see also the section on axiom K in the Agda Language Reference Manual.
- *exact-split* makes Agda accept only those definitions that behave like so-called *judgmental* equalities. See also the [Pattern matching and equality](#) section of the Agda Tools documentation.
- *safe* ensures that nothing is postulated outright—every non-MLTT axiom has to be an explicit assumption (e.g., an argument to a function or module); see also the `cmdoption-safe` section of the Agda Tools documentation and the [Safe Agda](#) section of the Agda Language Reference.

The `OPTIONS` pragma is usually followed by the start of a module and a list of import directives. For example, the collection of imports required for the present (`Demos.HSP`) module is relatively modest and is shown below.

```
{-# OPTIONS -without-K -exact-split -safe #-}
open import Algebras.Basic using ( 0 ; ℳ ; Signature )
```

```

module Demos.HSPnew {S : Signature 0 V} where

open import Agda.Primitive          using ( _⊔_ ; Isuc )
                                   renaming ( Set to Type )
open import Data.Product            using ( _×_ ;  $\Sigma$ -syntax ;  $\_,_;$  ;  $\Sigma$  )
                                   renaming ( proj1 to fst ; proj2 to snd )
open import Function               using ( id ; Surjection ; flip ; Injection ;  $\_ \circ \_$  )
                                   renaming ( Func to  $\_ \longrightarrow \_$  )
open import Level                  using ( Level )
open import Relation.Binary        using ( Setoid ; IsEquivalence ; Rel )
open import Relation.Binary.Definitions using ( Sym ; Symmetric ; Trans ; Transitive ; Reflexive )
open import Relation.Binary.PropositionalEquality
                                   using (  $\_ \equiv \_$  )
open import Relation.Unary         using ( Pred ;  $\_ \subseteq \_$  ;  $\_ \in \_$  )

import Function.Definitions         as FD
import Relation.Binary.PropositionalEquality as  $\equiv$ 
import Relation.Binary.Reasoning.Setoid as SetoidReasoning

open  $\_ \longrightarrow \_$  using ( cong ) renaming ( f to  $\_ \langle \$ \rangle \_$  )
open Setoid using ( Carrier ; isEquivalence )

```

Note, in particular, we prefer to use `Type` to denote the built-in `Set` type, and the infix long arrow symbol  `$\_ \longrightarrow \_$`  to denote the `Func` type of the standard library. We use `fst` and `snd` in place of `proj1` and `proj2` for the first and second projections out of the product type  `$\_ \times \_$`  and, when it improves readability of the code, we use the alternative notation `|_|` and `||_|` (resp.) for these projections.

## 2.2 Setoids

A *setoid* is a type packaged with an equivalence relation on the collection of inhabitants of that type. Setoids are useful for representing classical (set-theory-based) mathematics in a constructive, type-theoretic way because most mathematical structures are assumed to come equipped with some (often implicit) equivalence relation manifesting a notion of equality of elements, and therefore a type-theoretic representation of such a structure should also model its equality relation.

The `agda-algebras` library was first developed without the use of setoids, opting instead for specially constructed experimental quotient types. However, this approach resulted in code that was hard to comprehend and it became difficult to determine whether the resulting proofs were fully constructive. In particular, our initial proof of the Birkhoff variety theorem required postulating function extensionality, an axiom that is not provable in pure Martin-Löf type theory (MLTT). [reference needed]

In contrast, our current approach using setoids makes the equality relation of a given type explicit and this transparency can make it easier to determine the correctness and constructivity of the proofs. Using setoids we need no additional axioms beyond MLTT; in particular, no function extensionality axioms are postulated in our current formalization of Birkhoff's variety theorem.

### 2.3 Inverses of setoid functions

We define a data type that represent the semantic concept of the *image* of a function (cf. the `Overture.Func.Inverses` module of the `agda-algebras` library).

```
module _ {A : Setoid α ρa}{B : Setoid β ρb} where
  open Setoid B using ( _≈_ ; sym ) renaming ( Carrier to B )

  data Image_⊃_ (f : A → B) : B → Type (α ⊔ β ⊔ ρb) where
    eq : {b : B} → ∀ a → b ≈ (f $) a → Image f ⊃ b
```

An inhabitant of `Image f ⊃ b` is a dependent pair  $(a, p)$ , where  $a : A$  and  $p : b \approx f a$  is a proof that  $f$  maps  $a$  to  $b$ . Since the proof that  $b$  belongs to the image of  $f$  is always accompanied by a witness  $a : A$ , we can actually *compute* a (pseudo)inverse of  $f$ . For convenience, we define this inverse function, which we call `Inv`, and which takes an arbitrary  $b : B$  and a (witness, proof)-pair,  $(a, p) : \text{Image } f \ni b$ , and returns the witness  $a$ .

```
Inv : (f : A → B){b : B} → Image f ⊃ b → Carrier A
Inv _ (eq a _) = a

InvIsInverser : {f : A → B}{b : B}(q : Image f ⊃ b) → (f $) (Inv f q) ≈ b
InvIsInverser (eq _ p) = sym p
```

The latter (`InvIsInverser`) proves that `Inv f` is the range-restricted right-inverse of the setoid function  $f$ .

### 2.4 Injective and surjective setoid functions

If  $f : A \rightarrow B$  is a setoid function from  $A = (A, \approx_0)$  to  $B = (B, \approx_1)$ , then we call  $f$  *injective* provided  $\forall (a_0 a_1 : A), f \$ a_0 \approx_1 f \$ a_1$  implies  $a_0 \approx_0 a_1$ ; we call  $f$  *surjective* provided  $\forall (b : B), \exists (a : A)$  such that  $f \$ a \approx_1 b$ . We codify these definitions in Agda and prove some of their properties inside the next submodule where we first set the stage by declaring two setoids  $A$  and  $B$ , naming their equality relations, and making some definitions from the standard library available.

```
module _ {A : Setoid α ρa}{B : Setoid β ρb} where
  open Setoid A using () renaming ( _≈_ to _≈1_ )
  open Setoid B using () renaming ( _≈_ to _≈2_ )
  open FD _≈1_ _≈2_
```

The Agda Standard Library represents injective functions on bare types by the type `Injective`, which we now use to define `IsInjective` representing the property of being an injective setoid function. We then define the type `IsSurjective` to represent the property of being a surjective setoid function. Finally, we define `SurjInv` to represent the *right-inverse* of a surjective function. The definitions are as follows (cf. `Overture.Func.Injective` and `Overture.Func.Surjective` in the `agda-algebras` library).

```
IsInjective : (A → B) → Type (α ⊔ ρa ⊔ ρb)
IsInjective f = Injective ( _($)_ f )

IsSurjective : (A → B) → Type (α ⊔ β ⊔ ρb)
IsSurjective F = ∀ {y} → Image F ⊃ y
```

```

SurjInv : (f : A → B) → IsSurjective f → Carrier B → Carrier A
SurjInv f onto b = Inv f (onto {b})

```

### 2.4.1 Composition of injective and surjective setoid functions

Proving that the composition of injective setoid functions is again injective is simply a matter of composing the two assumed witnesses to injectivity. Proving that surjectivity is preserved under composition is only slightly more involved.

```

module _ {A : Setoid α ρa} {B : Setoid β ρb} {C : Setoid γ ρc}
  (f : A → B) (g : B → C) where

  o-IsInjective : IsInjective f → IsInjective g → IsInjective (g ∘ f)
  o-IsInjective finj ginj = finj ∘ ginj

  o-IsSurjective : IsSurjective f → IsSurjective g → IsSurjective (g ∘ f)
  o-IsSurjective onto onto {y} = Goal
  where
    mp : Image g ∋ y → Image g ∘ f ∋ y
    mp (eq c p) = η onto
    where
      open Setoid C using (trans)
      η : Image f ∋ c → Image g ∘ f ∋ y
      η (eq a q) = eq a (trans p (cong g q))

    Goal : Image g ∘ f ∋ y
    Goal = mp onto

```

## 2.5 Kernels

The *kernel* of a function  $f : A \rightarrow B$  (where  $A$  and  $B$  are bare types) is defined informally by

$$\{(x, y) \in A \times A : f\ x = f\ y\}.$$

This can be represented in Agda in a number of ways, but for our purposes we find it most convenient to define the kernel as an inhabitant of a (unary) predicate over the square of the function's domain.

```

module _ {A : Type α} {B : Type β} where

  kernel : Rel B ρ → (A → B) → Pred (A × A) ρ
  kernel _≈_ f (x, y) = f x ≈ f y

```

The kernel of a setoid function  $f : A \rightarrow B$  is defined informally by

$$\{(x, y) \in A \times A : f\ \$\ x \approx f\ \$\ y\},$$

where  $\_ \approx \_$  denotes the equality of  $B$ .

```

module _ {A : Setoid α ρa} {B : Setoid β ρb} where
  open Setoid A using () renaming (Carrier to A)

  ker : (A → B) → Pred (A × A) ρb
  ker g (x, y) = g ($) x ≈ g ($) y where open Setoid B using (_≈_)

```

### 3 Algebras

In this section we define the notion of an algebraic structure whose *domain* (or “carrier” or “universe”) is a setoid. Our first goal is to develop a working vocabulary and formal types for classical (single-sorted, set-based) universal algebra.

#### 3.1 Signature of an algebra

In model theory, the *signature*  $S = (C, F, R, \rho)$  of a structure consists of three (possibly empty) sets  $C$ ,  $F$ , and  $R$ —called *constant symbols*, *function symbols*, and *relation symbols*, respectively—along with a function  $\rho : C + F + R \rightarrow \mathbb{N}$  that assigns an *arity* to each symbol. Often, but not always,  $\mathbb{N}$  is taken to be the set of natural numbers.

As our focus here is universal algebra, we are more concerned with the restricted notion of an *algebraic signature* (or *signature* for algebraic structures), by which we mean a pair  $S = ((F, \rho))$  consisting of a collection  $F$  of *operation symbols* and an *arity function*  $\rho : F \rightarrow \mathbb{N}$  that maps each operation symbol to its arity; here,  $\mathbb{N}$  denotes the *arity type*. Heuristically, the arity  $\rho f$  of an operation symbol  $f \in F$  may be thought of as the “number of arguments” that  $f$  takes as “input.”

If the arity of  $f$  is  $n$ , then we call  $f$  an  $n$ -*ary* operation symbol. In case  $n$  is 0 (or 1 or 2 or 3, respectively) we call the function *nullary* (or *unary* or *binary* or *ternary*, respectively).

If  $A$  is a set and  $f$  is a  $(\rho f)$ -ary operation on  $A$  then the arguments of  $f$  form a  $(\rho f)$ -tuple, say,  $(a_0, a_1, \dots, a_{(\rho f) - 1})$ , which may be viewed as the graph of a function, say,  $a : \rho f \rightarrow A$ . When the codomain of  $\rho$  is  $\mathbb{N}$ , we may view  $\rho f$  as the finite set  $\{0, 1, \dots, \rho f - 1\}$ . Thus, by identifying the  $\rho f$ -th power of  $A$  with the type  $\rho f \rightarrow A$  of functions from  $\{0, 1, \dots, \rho f - 1\}$  to  $A$ , we identify the collection of all tuples of arguments of  $f$  with the function type  $(\rho f \rightarrow A) \rightarrow A$ .

##### 3.1.1 Signature type

The `agda-algebras` library represents a *signature* as an inhabitant of the following dependent pair type.

$$\begin{aligned} \text{Signature} &: (\mathbb{O} \mathcal{V} : \text{Level}) \rightarrow \text{Type} (\text{Isuc } (\mathbb{O} \sqcup \mathcal{V})) \\ \text{Signature } \mathbb{O} \mathcal{V} &= \Sigma [ F \in \text{Type } \mathbb{O} ] (F \rightarrow \text{Type } \mathcal{V}) \end{aligned}$$

Using special syntax for the first and second projections— $| \_ |$  and  $\| \_ \|$  (resp.)—if  $S : \text{Signature } \mathbb{O} \mathcal{V}$  is a signature, then

- $| S |$  denotes the set of operation symbols;
- $\| S \|$  denotes the arity function.

Thus, if  $f : | S |$  is an operation symbol in the signature  $S$ , then  $\| S \| f$  is the arity of  $f$ .

We need to augment the ordinary `Signature` type so that it supports algebras over setoid domains. To do so, we define an operator  $\langle \_ \rangle$  which translates an ordinary signature into a setoid signature, that is, a signature over a setoid domain. But first we must resolve a technical issue involving dependent types that we now describe.

Suppose we are given two operations  $f$  and  $g$  and we have a tuple of arguments for  $f$ , say,  $u : \| S \| f \rightarrow A$ , and a tuple of arguments for  $g$ , say,  $v : \| S \| g \rightarrow A$ . If we know that  $f$  is identically equal to  $g$ —that is,  $f \equiv g$  (intensionally)—then we should be able to check whether  $u$  and  $v$  are pointwise equal. The problem here is that  $u$  and  $v$  ostensibly inhabit different types. To compare  $u$  and  $v$  we must convince Agda that, from  $f \equiv g$  we can deduce

that  $u$  and  $v$  are actually of the same type. The type `EqArgs` (defined below, and adapted from Andreas Abel’s development [ref needed]) resolves this technical issue nicely.

```
EqArgs : {S : Signature ⊗ V} {ξ : Setoid α ρa}
  → ∀ {f g} → f ≡ g → (|| S || f → Carrier ξ) → (|| S || g → Carrier ξ) → Type (V ⊔ ρa)

EqArgs {ξ = ξ} ≡ .refl u v = ∀ i → u i ≈ v i where open Setoid ξ using ( _≈_ )
```

Now we are ready to define the `<_>` operator, which translates an ordinary signature into a setoid signature.

```
module _ where
  open Setoid using ( _≈_ )
  open IsEquivalence using ( refl ; sym ; trans )

  <_> : Signature ⊗ V → Setoid α ρa → Setoid _ _

  Carrier (< S > ξ) = Σ[ f ∈ | S | ] ((|| S || f) → ξ .Carrier)

  _≈_ (< S > ξ) (f , u) (g , v) = Σ[ eqv ∈ f ≡ g ] EqArgs{ξ = ξ} eqv u v

  refl (isEquivalence (< S > ξ)) = ≡.refl , λ i → Setoid.refl ξ
  sym (isEquivalence (< S > ξ)) (≡.refl , g) = ≡.refl , λ i → Setoid.sym ξ (g i)
  trans (isEquivalence (< S > ξ)) (≡.refl , g) (≡.refl , h) = ≡.refl , λ i → Setoid.trans ξ (g i) (h i)
```

## 3.2 Algebra type

Informally, an *algebraic structure in the signature*  $S = (F, \rho)$  (or *S-algebra*) is typically denoted by  $\mathbf{A} = (A, F^A)$  and consists of

- $A$  := a *nonempty* set (or type), called the *domain* (or *carrier* or *universe*) of the algebra;
- $F^A$  :=  $\{ f^A \mid f \in F, f^A : (\rho f \rightarrow A) \rightarrow A \}$ , a collection of *operations* on  $A$ ;
- a (potentially empty) collection of *identities* satisfied by elements and operations of  $A$ .

We represent an algebra in Agda using a record type with two fields:

- `Domain` is a setoid denoting the underlying *universe* of the algebra (informally, the set of elements of the algebra);
- `Interp` represents the *interpretation* in the algebra of each operation symbol of the given signature. The record type `Func` from the Agda Standard Library provides what we need for an operation on the domain setoid.

Let us present the definition of the `Algebra` type and then discuss the definition of the `Func` type that provides the interpretation of each operation symbol.

```
record Algebra α ρ : Type (⊗ ⊔ V ⊔ Isuc (α ⊔ ρ)) where
  field Domain : Setoid α ρ
  Interp : (< S > Domain) → Domain
```

The `Interp` field actually has type `Func (< S > Domain) Domain` (recall we renamed `Func` as the infix long-arrow symbol). The `Func` type is from the standard library and is defined as a record type with two fields. In the present instance, the fields are

- a function  $f : \text{Carrier } (< S > \text{Domain}) \rightarrow \text{Carrier Domain}$

- a proof `cong : f Preserves  $\approx_1$   $\rightarrow$   $\approx_2$`  that `f` preserves the underlying setoid equalities.

Thus `Interp` gives, for each operation symbol in the signature  $S$ , a setoid function `f`—namely, a function where the domain is a power of `Domain` and the codomain is `Domain`—along with a proof that all operations so interpreted respect the underlying setoid equality on `Domain`.

Next we define some syntactic sugar that will make our Agda code easier to read and comprehend. If  $\mathbf{A}$  is an algebra, then

- `D[  $\mathbf{A}$  ]` will denote the setoid `Domain  $\mathbf{A}$` , and
- `U[  $\mathbf{A}$  ]` will be the underlying carrier or “universe” of the algebra  $\mathbf{A}$ .
- `f ^  $\mathbf{A}$`  will denote the interpretation in the algebra  $\mathbf{A}$  of the operation symbol `f`,

```
open Algebra
```

```
D[_] : Algebra  $\alpha$   $\rho^a$   $\rightarrow$  Setoid  $\alpha$   $\rho^a$ 
```

```
D[  $\mathbf{A}$  ] = Domain  $\mathbf{A}$ 
```

```
U[_] : Algebra  $\alpha$   $\rho^a$   $\rightarrow$  Type  $\alpha$ 
```

```
U[  $\mathbf{A}$  ] = Carrier (Domain  $\mathbf{A}$ )
```

```
_ ^ _ : (f : | S |) ( $\mathbf{A}$  : Algebra  $\alpha$   $\rho^a$ )  $\rightarrow$  (| S | f  $\rightarrow$  U[  $\mathbf{A}$  ])  $\rightarrow$  U[  $\mathbf{A}$  ]
```

```
f ^  $\mathbf{A}$  =  $\lambda$  a  $\rightarrow$  (Interp  $\mathbf{A}$ ) ($$) (f , a)
```

### 3.3 Universe lifting of algebra types

A technical aspect of dealing with the noncumulativity of the hierarchy of type levels in Agda...

```
module _ ( $\mathbf{A}$  : Algebra  $\alpha$   $\rho^a$ ) where
```

```
open Setoid D[  $\mathbf{A}$  ] using (  $\approx$  ; refl ; sym ; trans )
```

```
open Level
```

```
Lift-Algl : ( $\ell$  : Level)  $\rightarrow$  Algebra ( $\alpha \sqcup \ell$ )  $\rho^a$ 
```

```
Domain (Lift-Algl  $\ell$ ) =
```

```
record { Carrier = Lift  $\ell$  U[  $\mathbf{A}$  ]
```

```
;  $\approx$  =  $\lambda$  x y  $\rightarrow$  lower x  $\approx$  lower y
```

```
; isEquivalence = record { refl = refl ; sym = sym ; trans = trans } }
```

```
Interp (Lift-Algl  $\ell$ ) ($$) (f , la) = lift ((f ^  $\mathbf{A}$ ) (lower  $\circ$  la))
```

```
cong (Interp (Lift-Algl  $\ell$ )) ( $\equiv$ .refl , la=lb) = cong (Interp  $\mathbf{A}$ ) ( $\equiv$ .refl , la=lb))
```

```
Lift-Algr : ( $\ell$  : Level)  $\rightarrow$  Algebra  $\alpha$  ( $\rho^a \sqcup \ell$ )
```

```
Domain (Lift-Algr  $\ell$ ) =
```

```
record { Carrier = U[  $\mathbf{A}$  ]
```

```
;  $\approx$  =  $\lambda$  x y  $\rightarrow$  Lift  $\ell$  (x  $\approx$  y)
```

```
; isEquivalence = record { refl = lift refl
```

```
; sym =  $\lambda$  x  $\rightarrow$  lift (sym (lower x))
```

```
; trans =  $\lambda$  x y  $\rightarrow$  lift (trans (lower x) (lower y)) } }
```

```
Interp (Lift-Algr  $\ell$ ) ($$) (f , la) = (f ^  $\mathbf{A}$ ) la
```

```
cong (Interp (Lift-Algr  $\ell$ )) ( $\equiv$ .refl , la=lb) = lift (cong (Interp  $\mathbf{A}$ ) ( $\equiv$ .refl ,  $\lambda$  i  $\rightarrow$  lower (la=lb i)))
```

```
Lift-Alg : ( $\mathbf{A}$  : Algebra  $\alpha$   $\rho^a$ ) ( $\ell_0$   $\ell_1$  : Level)  $\rightarrow$  Algebra ( $\alpha \sqcup \ell_0$ ) ( $\rho^a \sqcup \ell_1$ )
```

```
Lift-Alg  $\mathbf{A}$   $\ell_0$   $\ell_1$  = Lift-Algr (Lift-Algl  $\mathbf{A}$   $\ell_0$ )  $\ell_1$ 
```



### 3.4 Product Algebras

(cf. the `Algebras.Func.Products` module of the `agda-algebras` library.)

```

module _ {ℓ : Level}{I : Type ℓ} where

  ⊔ : (A : I → Algebra α ρa) → Algebra (α ⊔ ℓ) (ρa ⊔ ℓ)
  Domain (⊔ A) =
    record { Carrier = ∀ i → ⋃[ A i ]
          ; _≈_ = λ a b → ∀ i → (Setoid._≈_ D[ A i ]) (a i)(b i)
          ; isEquivalence =
              record { refl = λ i → IsEquivalence.refl (isEquivalence D[ A i ])
                    ; sym = λ x i → IsEquivalence.sym (isEquivalence D[ A i ])(x i)
                    ; trans = λ x y i → IsEquivalence.trans (isEquivalence D[ A i ])(x i)(y i) }}
  Interp (⊔ A) ($) (f , a) = λ i → (f ^ (A i)) (flip a i)
  cong (Interp (⊔ A)) (≡.refl , f=g) = λ i → cong (Interp (A i)) (≡.refl , flip f=g i)

```

## 4 Homomorphisms

### 4.1 Basic definitions

Here are some useful definitions and theorems extracted from the `Homomorphisms.Func.Basic` module of the `agda-algebras` library.

```

module _ (A : Algebra α ρa)(B : Algebra β ρb) where
  private ov = ⅉ ⊔ ℳ ; a = α ⊔ ρa ; b = β ⊔ ρb ; c = ⅉ ⊔ ℳ ⊔ α ⊔ ρa ⊔ β ⊔ ρb

  compatible-map-op : (D[ A ] → D[ B ]) → | S | → Type (ℳ ⊔ α ⊔ ρb)
  compatible-map-op h f = ∀ {a} → (h ($) ((f ^ A) a)) ≈2 ((f ^ B) (λ x → (h ($) (a x))))
    where open Setoid D[ B ] using () renaming ( _≈_ to _≈2_ )

  compatible-map : (D[ A ] → D[ B ]) → Type (ov ⊔ α ⊔ ρb)
  compatible-map h = ∀ {f} → compatible-map-op h f

  – The property of being a homomorphism.
  record IsHom (h : D[ A ] → D[ B ]) : Type (ov ⊔ α ⊔ ρb) where
    constructor mkhom
    field compatible : compatible-map h

  – The type of homomorphisms.
  hom : Type c
  hom = Σ (D[ A ] → D[ B ]) IsHom

```

### 4.2 Monomorphisms and epimorphisms

```

record IsMon (h : D[ A ] → D[ B ]) : Type (ov ⊔ a ⊔ ρb) where
  field isHom : IsHom h
  isInjective : IsInjective h

  HomReduct : hom
  HomReduct = h , isHom

  mon : Type c
  mon = Σ (D[ A ] → D[ B ]) IsMon

```

```

record IsEpi (h :  $\mathbb{D}[A] \rightarrow \mathbb{D}[B]$ ) : Type (ov  $\sqcup \alpha \sqcup b$ ) where
  field isHom : IsHom h
  isSurjective : IsSurjective h

HomReduct : hom
HomReduct = h , isHom

epi : Type c
epi =  $\Sigma (\mathbb{D}[A] \rightarrow \mathbb{D}[B])$  IsEpi

open IsHom ; open IsMon ; open IsEpi

module _ (A : Algebra  $\alpha \rho^a$ ) (B : Algebra  $\beta \rho^b$ ) where

  mon $\rightarrow$ intohom : mon A B  $\rightarrow \Sigma [h \in \text{hom A B}]$  IsInjective | h |
  mon $\rightarrow$ intohom (hh , hhM) = (hh , isHom hhM) , isInjective hhM

  epi $\rightarrow$ ontohom : epi A B  $\rightarrow \Sigma [h \in \text{hom A B}]$  IsSurjective | h |
  epi $\rightarrow$ ontohom (hh , hhE) = (hh , isHom hhE) , isSurjective hhE

```

### 4.3 Basic properties of homomorphisms

Here are some definitions and theorems extracted from the `Homomorphisms.Func.Properties` module of the `agda-algebras` library.

#### 4.3.1 Composition of homomorphisms

The composition of homomorphisms is again a homomorphism. Similarly, the composition of epimorphisms is again an epimorphism.

```

module _ {A : Algebra  $\alpha \rho^a$ } {B : Algebra  $\beta \rho^b$ } {C : Algebra  $\gamma \rho^c$ }
  {g :  $\mathbb{D}[A] \rightarrow \mathbb{D}[B]$ } {h :  $\mathbb{D}[B] \rightarrow \mathbb{D}[C]$ } where

  open Setoid  $\mathbb{D}[C]$  using ( trans )

  o-is-hom : IsHom A B g  $\rightarrow$  IsHom B C h  $\rightarrow$  IsHom A C (h  $\langle \circ \rangle$  g)
  o-is-hom ghom hhom = mkhom c
  where
    c : compatible-map A C (h  $\langle \circ \rangle$  g)
    c = trans (cong h (compatible ghom)) (compatible hhom)

  o-is-epi : IsEpi A B g  $\rightarrow$  IsEpi B C h  $\rightarrow$  IsEpi A C (h  $\langle \circ \rangle$  g)
  o-is-epi gE hE = record { isHom = o-is-hom (isHom gE) (isHom hE)
    ; isSurjective = o-IsSurjective g h (isSurjective gE) (isSurjective hE) }

module _ {A : Algebra  $\alpha \rho^a$ } {B : Algebra  $\beta \rho^b$ } {C : Algebra  $\gamma \rho^c$ } where

  o-hom : hom A B  $\rightarrow$  hom B C  $\rightarrow$  hom A C
  o-hom (h , hhom) (g , ghom) = (g  $\langle \circ \rangle$  h) , o-is-hom hhom ghom

  o-epi : epi A B  $\rightarrow$  epi B C  $\rightarrow$  epi A C
  o-epi (h , hepi) (g , gepi) = (g  $\langle \circ \rangle$  h) , o-is-epi hepi gepi

```

### 4.3.2 Universe lifting of homomorphisms

First we define the identity homomorphism for setoid algebras and then we prove that the operations of lifting and lowering of a setoid algebra are homomorphisms.

## 4.4 Factorization of homomorphisms

If  $g : \text{hom } A \ B$ ,  $h : \text{hom } A \ C$ ,  $h$  is surjective, and  $\ker h \subseteq \ker g$ , then there exists  $\varphi : \text{hom } C \ B$  such that  $g = \varphi \circ h$  (cf. the `Homomorphisms.Func.Factor` module of the `agda-algebras` library).

```

module _ {A : Algebra α ρa}{B : Algebra β ρb}{C : Algebra γ ρc}
  (gh : hom A B)(hh : hom A C) where
  open Setoid ℙ[ B ] using ( renaming ( _≈_ to _≈2_ ; sym to sym2 ) )
  open Setoid ℙ[ C ] using ( trans ) renaming ( _≈_ to _≈3_ ; sym to sym3 )
  open SetoidReasoning ℙ[ B ]
  private gfunc = | gh | ; g = _⟨$⟩_ gfunc ; hfunc = | hh | ; h = _⟨$⟩_ hfunc

  HomFactor : kernel _≈3_ h ⊆ kernel _≈2_ g → IsSurjective hfunc
    → ∑[ φ ∈ hom C B ] ∀ a → (g a) ≈2 | φ | ⟨$⟩ (h a)

  HomFactor Khg hE = (φmap , φhom) , gφh
  where
    kerpres : ∀ a0 a1 → h a0 ≈3 h a1 → g a0 ≈2 g a1
    kerpres a0 a1 hyp = Khg hyp

    h-1 : ℙ[ C ] → ℙ[ A ]
    h-1 = SurjInv hfunc hE

    η : ∀ {c} → h (h-1 c) ≈3 c
    η = InvIsInverser hE

    ζ : ∀ {x y} → x ≈3 y → h (h-1 x) ≈3 h (h-1 y)
    ζ xy = trans η (trans xy (sym3 η))

    φmap : ℙ[ C ] → ℙ[ B ]
    _⟨$⟩_ φmap = g ∘ h-1
    cong φmap = Khg ∘ ζ
    open _→_ φmap using ( renaming (cong to φcong) )

    gφh : (a : ℙ[ A ]) → g a ≈2 φmap ⟨$⟩ (h a)
    gφh a = Khg (sym3 η)

    φcomp : compatible-map C B φmap
    φcomp {f}{c} =
      begin
        φmap ⟨$⟩ ((f ^ C) c) ≈~ ⟨ φcong (cong (Interp C) (≡.refl , λ _ → η)) ⟩
        g (h-1 ((f ^ C)(h ∘ h-1 ∘ c))) ≈~ ⟨ φcong (compatible || hh ||) ⟩
        g (h-1 (h ((f ^ A)(h-1 ∘ c)))) ≈~ ⟨ gφh ((f ^ A)(h-1 ∘ c)) ⟩
        g ((f ^ A)(h-1 ∘ c)) ≈~ ⟨ compatible || gh || ⟩
        (f ^ B)(g ∘ (h-1 ∘ c)) ■

    φhom : IsHom C B φmap
    compatible φhom = φcomp

```

### 4.5 Isomorphisms

(cf. the `Homomorphisms.Func.Isomorphisms` of the `agda-algebras` library.)

Two structures are *isomorphic* provided there are homomorphisms going back and forth between them which compose to the identity map.

```

module _ (A : Algebra α ρa) (B : Algebra β ρb) where
  open Setoid ℙ[ A ] using ( _≈_ ; sym ; trans )
  open Setoid ℙ[ B ] using () renaming ( _≈_ to _≈b_ ; sym to symb ; trans to transb )

  record _≅_ : Type (ℳ ⊔ ℳ ⊔ α ⊔ β ⊔ ρa ⊔ ρb) where
    constructor mkiso
    field
      to : hom A B
      from : hom B A
      to~from : ∀ b → (| to | ⟨$⟩ (| from | ⟨$⟩ b)) ≈b b
      from~to : ∀ a → (| from | ⟨$⟩ (| to | ⟨$⟩ a)) ≈ a

  toIsSurjective : IsSurjective | to |
  toIsSurjective {y} = eq (| from | ⟨$⟩ y) (symb (to~from y))

  toIsInjective : IsInjective | to |
  toIsInjective {x} {y} xy = Goal
    where
      ξ : | from | ⟨$⟩ (| to | ⟨$⟩ x) ≈ | from | ⟨$⟩ (| to | ⟨$⟩ y)
      ξ = cong | from | xy
      Goal : x ≈ y
      Goal = trans (sym (from~to x)) (trans ξ (from~to y))

  fromIsSurjective : IsSurjective | from |
  fromIsSurjective {y} = eq (| to | ⟨$⟩ y) (sym (from~to y))

  fromIsInjective : IsInjective | from |
  fromIsInjective {x} {y} xy = Goal
    where
      ξ : | to | ⟨$⟩ (| from | ⟨$⟩ x) ≈b | to | ⟨$⟩ (| from | ⟨$⟩ y)
      ξ = cong | to | xy
      Goal : x ≈b y
      Goal = transb (symb (to~from x)) (transb ξ (to~from y))

  open _≅_

```

#### 4.5.1 Properties of isomorphisms

```

≅-refl : Reflexive ( _≅_ {α}{ρa} )
≅-refl {α}{ρa}{A} = mkiso id id (λ b → refl) λ a → refl
  where open Setoid ℙ[ A ] using ( refl )

≅-sym : Sym ( _≅_ {β}{ρb} ) ( _≅_ {α}{ρa} )
≅-sym φ = mkiso (from φ) (to φ) (from~to φ) (to~from φ)

≅-trans : Trans ( _≅_ {α}{ρa} ) ( _≅_ {β}{ρb} ) ( _≅_ {α}{ρa}{γ}{ρc} )
≅-trans {ρc = ρc}{A}{B}{C} ab bc = mkiso f g τ ν
  where

```

```

open Setoid [ A ] using ( _≈_ ; trans )
open Setoid [ C ] using () renaming ( _≈_ to _≈c_ ; trans to transc )
f : hom A C
f = o-hom (to ab) (to bc)
g : hom C A
g = o-hom (from bc) (from ab)
τ : ∀ b → (| f | ⟨$⟩ (| g | ⟨$⟩ b)) ≈c b
τ b = transc (cong | to bc | (to~from ab (| from bc | ⟨$⟩ b))) (to~from bc b)
ν : ∀ a → (| g | ⟨$⟩ (| f | ⟨$⟩ a)) ≈ a
ν a = trans (cong | from ab | (from~to bc (| to ab | ⟨$⟩ a))) (from~to ab a)

```

Fortunately, the lift operation preserves isomorphism (i.e., it's an *algebraic invariant*). As our focus is universal algebra, this is important and is what makes the lift operation a workable solution to the technical problems that arise from the noncumulativity of Agda's universe hierarchy.

## 4.6 Homomorphic Images

We begin with what for our purposes is the most useful way to represent the class of *homomorphic images* of an algebra in dependent type theory (cf. the [Homomorphisms.Func.HomomorphicImages](#) module of the `agda-algebras` library). (The first definition is merely a short-hand.)

```

ov : Level → Level
ov α = 0 ⊔ ℳ ⊔ lsuc α

_!sHomImageOf_ : (B : Algebra β ρb)(A : Algebra α ρa) → Type (0 ⊔ ℳ ⊔ α ⊔ β ⊔ ρa ⊔ ρb)
B !sHomImageOf A = Σ[ φ ∈ hom A B ] !sSurjective | φ |

HomImages : Algebra α ρa → Type (α ⊔ ρa ⊔ ov (β ⊔ ρb))
HomImages {β = β}{ρb = ρb} A = Σ[ B ∈ Algebra β ρb ] B !sHomImageOf A

```

These types should be self-explanatory, but just to be sure, let's describe the Sigma type appearing in the second definition. Given an  $S$ -algebra  $A : \text{Algebra } \alpha \rho$ , the type `HomImages A` denotes the class of algebras  $B : \text{Algebra } \beta \rho$  with a map  $\varphi : |A| \rightarrow |B|$  such that  $\varphi$  is a surjective homomorphism.

## 5 Subalgebras

### 5.1 Basic definitions

```

_≤_ : Algebra α ρa → Algebra β ρb → Type (0 ⊔ ℳ ⊔ α ⊔ ρa ⊔ β ⊔ ρb)
A ≤ B = Σ[ h ∈ hom A B ] !sInjective | h |

```

### 5.2 Basic properties

```

≤-reflexive : {A : Algebra α ρa} → A ≤ A
≤-reflexive {A = A} = id , id

mon→≤ : {A : Algebra α ρa}{B : Algebra β ρb} → mon A B → A ≤ B
mon→≤ {A = A}{B} x = mon→intohom A B x

module _ {A : Algebra α ρa}{B : Algebra β ρb}{C : Algebra γ ρc} where

```

```

≤-trans : A ≤ B → B ≤ C → A ≤ C
≤-trans ( f , finj ) ( g , ginj ) = ( o-hom f g ) , o-IsInjective | f | g | finj ginj

≅-trans-≤ : A ≅ B → B ≤ C → A ≤ C
≅-trans-≤ A≅B ( h , hinj ) = ( o-hom ( to A≅B ) h ) , ( o-IsInjective | to A≅B | | h | ( toIsInjective A≅B ) hinj )

```

### 5.3 Products of subalgebras

```

module _ {ι : Level} {I : Type ι} {A : I → Algebra α ρa} {B : I → Algebra β ρb} where

  ⌊-≤ : (∀ i → B i ≤ A i) → ⌊ B ≤ ⌊ A
  ⌊-≤ B≤A = (hfunc , hhom) , hM
  where
    hi : ∀ i → hom (B i) (A i)
    hi i = | B≤A i |

    hfunc : D[ ⌊ B ] → D[ ⌊ A ]
    (hfunc ⟨$⟩ x) i = | hi i | ⟨$⟩ (x i)
    cong hfunc = λ xy i → cong | hi i | (xy i)

    hhom : IsHom (⌊ B) (⌊ A) hfunc
    compatible hhom = λ i → compatible || hi i ||

    hM : IsInjective hfunc
    hM = λ xy i → || B≤A i || (xy i)

```

## 6 Terms

### 6.1 Basic definitions

Fix a signature  $S$  and let  $X$  denote an arbitrary nonempty collection of variable symbols. Assume the symbols in  $X$  are distinct from the operation symbols of  $S$ , that is  $X \cap |S| = \emptyset$ .

By a *word* in the language of  $S$ , we mean a nonempty, finite sequence of members of  $X \cup |S|$ . We denote the concatenation of such sequences by simple juxtaposition.

Let  $S_0$  denote the set of nullary operation symbols of  $S$ . We define by induction on  $n$  the sets  $T_n$  of *words* over  $X \cup |S|$  as follows (cf. [1, Def. 4.19]):

1.  $T_0 := X \cup S_0$ , and
2.  $T_{n+1} := T_n \cup \mathcal{T}_n$ .

where  $\mathcal{T}_n$  is the collection of all  $f \ t$  such that  $f : |S|$  and  $t : ||S|| f \rightarrow T_n$ . (Recall,  $||S|| f$  is the arity of the operation symbol  $f$ .)

We define the collection of *terms* in the signature  $S$  over  $X$  by  $\text{Term } X := \bigcup_n T_n$ . By an  $S$ -*term* we mean a term in the language of  $S$ .

The definition of  $\text{Term } X$  is recursive, indicating that an inductive type could be used to represent the semantic notion of terms in type theory. Indeed, such a representation is given by the following inductive type.

```

data Term (X : Type χ) : Type (ov χ) where
  g : X → Term X
  node : (f : |S|)(t : ||S|| f → Term X) → Term X
open Term

```

This is a very basic inductive type that represents each term as a tree with an operation symbol at each **node** and a variable symbol at each leaf ( $\mathcal{g}$ ); hence the constructor names ( $\mathcal{g}$  for “generator” and **node** for “node”).

**Notation.** As usual, the type  $X$  represents an arbitrary collection of variable symbols. Recall,  $\text{ov } \chi$  is our shorthand notation for the universe level  $\mathcal{O} \sqcup \mathcal{V} \sqcup \text{Isuc } \chi$ .

## 6.2 Equality of terms

We take a different approach here, using Setoids instead of quotient types. That is, we will define the collection of terms in a signature as a setoid with a particular equality-of-terms relation, which we must define. Ultimately we will use this to define the (absolutely free) term algebra as a Algebra whose carrier is the setoid of terms.

```
module _ {X : Type  $\chi$ } where
  data  $\dot{=}$  : Term X  $\rightarrow$  Term X  $\rightarrow$  Type (ov  $\chi$ ) where
    rfl : {x y : X}  $\rightarrow$  x  $\equiv$  y  $\rightarrow$  ( $\mathcal{g}$  x)  $\dot{=}$  ( $\mathcal{g}$  y)
    gnl :  $\forall$  {f} {s t :  $\parallel S \parallel$  f  $\rightarrow$  Term X}  $\rightarrow$  ( $\forall$  i  $\rightarrow$  (s i)  $\dot{=}$  (t i))  $\rightarrow$  (node f s)  $\dot{=}$  (node f t)
```

It is easy to show that the equality-of-terms relation  $\dot{=}$  is an equivalence relation, so we omit the formal proof. (See the `Terms.Func.Basic` module of the `agda-algebras` library for details.)

## 6.3 The term algebra

For a given signature  $S$ , if the type `Term X` is nonempty (equivalently, if  $X$  or  $\parallel S \parallel$  is nonempty), then we can define an algebraic structure, denoted by  $\mathbf{T} X$  and called the *term algebra in the signature  $S$  over  $X$* . Terms are viewed as acting on other terms, so both the domain and basic operations of the algebra are the terms themselves.

- For each operation symbol  $f : \parallel S \parallel$ , denote by  $f^\wedge(\mathbf{T} X)$  the operation on `Term X` that maps a tuple  $t : \parallel S \parallel$   $f \rightarrow \mathbf{T} X$  to the formal term  $f t$ .
- Define  $\mathbf{T} X$  to be the algebra with universe  $\parallel \mathbf{T} X \parallel := \text{Term } X$  and operations  $f^\wedge(\mathbf{T} X)$ , one for each symbol  $f$  in  $\parallel S \parallel$ .

In Agda the term algebra can be defined as simply as one might hope.

```
TermSetoid : (X : Type  $\chi$ )  $\rightarrow$  Setoid (ov  $\chi$ ) (ov  $\chi$ )
TermSetoid X = record { Carrier = Term X ;  $\approx$  =  $\dot{=}$  ; isEquivalence =  $\dot{=}$ -isEquiv }

T : (X : Type  $\chi$ )  $\rightarrow$  Algebra (ov  $\chi$ ) (ov  $\chi$ )
Algebra.Domain (T X) = TermSetoid X
Algebra.Interp (T X) ($) (f , ts) = node f ts
cong (Algebra.Interp (T X)) ( $\equiv$ .refl , ss $\dot{=}$ ts) = gnl ss $\dot{=}$ ts
```

## 6.4 Interpretation of terms

The approach to terms and their interpretation in this module was inspired by Andreas Abel’s formal proof of Birkhoff’s completeness theorem.<sup>1</sup>

<sup>1</sup> See <http://www.cse.chalmers.se/~abela/agda/MultiSortedAlgebra.pdf>.

A substitution from  $X$  to  $Y$  associates a term in  $X$  with each variable in  $Y$ . The definition of `Sub` given here is essentially the same as the one given by Andreas Abel, as is the recursive definition of the syntax `t [σ]`, which denotes a term `t` applied to a substitution `σ`.

```
Sub : Type χ → Type χ → Type (ov χ)
Sub X Y = (y : Y) → Term X

_[] : {X Y : Type χ} (t : Term Y) (σ : Sub X Y) → Term X
(g x) [σ] = σ x
(node f ts) [σ] = node f (λ i → ts i [σ])
```

An environment for an algebra  $\mathbf{A}$  in a context  $X$  is a map that assigns to each variable  $x : X$  an element in the domain of  $\mathbf{A}$ , packaged together with an equality of environments, which is simply pointwise equality (relatively to the setoid equality of the underlying domain of  $\mathbf{A}$ ).

```
module Environment (A : Algebra α ℓ) where
  open Setoid D[ A ] using ( _≈_ ; refl ; sym ; trans )
  Env : Type χ → Setoid _ _
  Env X = record { Carrier = X → U[ A ]
                  ; _≈_ = λ ρ ρ' → (x : X) → ρ x ≈ ρ' x
                  ; isEquivalence = record { refl = λ _ → refl
                                              ; sym = λ h x → sym (h x)
                                              ; trans = λ g h x → trans (g x)(h x) }}
```

Next we define *evaluation of a term* in an environment  $\rho$ , interpreted in the algebra  $\mathbf{A}$ .

```
[] : {X : Type χ} (t : Term X) → (Env X) → D[ A ]
[g x] ($) ρ = ρ x
[node f args] ($) ρ = (Interp A) ($) (f , λ i → [args i] ($) ρ)
cong [g x] u ≈ v = u ≈ v x
cong [node f args] x ≈ y = cong (Interp A) (≡.refl , λ i → cong [args i] x ≈ y)
```

An equality between two terms holds in a model if the two terms are equal under all valuations of their free variables (cf. Andreas Abel's formal proof of Birkhoff's completeness theorem).

```
Equal : ∀ {X : Type χ} (s t : Term X) → Type _
Equal {X = X} s t = ∀ (ρ : Carrier (Env X)) → [s] ($) ρ ≈ [t] ($) ρ

⇒Equal : {X : Type χ} (s t : Term X) → s ⇒ t → Equal s t
⇒Equal .(g _) .(g _) (rfl ≡.refl) = λ _ → refl
⇒Equal (node _ s) (node _ t) (gnt x) =
  λ ρ → cong (Interp A) (≡.refl , λ i → ⇒Equal (s i) (t i) (x i) ρ)
```

The proof that `Equal` is an equivalence relation is trivial, so we omit it. (See the `Varieties.Func.SoundAndComplete` module of the `agda-algebras` library for details.)

Evaluation of a substitution gives an environment (cf. [Andreas Abel's formal proof of Birkhoff's completeness theorem](<http://www.cse.chalmers.se/~abela/agda/MultiSortedAlgebra.pdf>)).



```

[ ]s : {X Y : Type χ} → Sub X Y → Carrier (Env X) → Carrier (Env Y)
[ σ ]s ρ x = [ σ x ] ($) ρ

```

Next we prove that  $\llbracket t \llbracket \sigma \rrbracket \rrbracket \rho \simeq \llbracket t \rrbracket \llbracket \sigma \rrbracket \rho$  (cf. Andreas Abel’s formal proof of Birkhoff’s completeness theorem).

```

substitution : {X Y : Type χ} → (t : Term Y) (σ : Sub X Y) (ρ : Carrier (Env X))
  → [ t [ σ ] ] ($) ρ ≈ [ t ] ($) ([ σ ]s ρ)

substitution (g x) σ ρ = refl
substitution (node f ts) σ ρ = cong (Interp A)(≡.refl , λ i → substitution (ts i) σ ρ)

```

## 6.5 Compatibility of terms

We now prove two important facts about term operations. The first of these, which is used very often in the sequel, asserts that every term commutes with every homomorphism.

```

module _ {X : Type χ} {A : Algebra α ρa} {B : Algebra β ρb} (hh : hom A B) where
  open Setoid D[ B ] using ( _≈_ ; refl )
  open SetoidReasoning D[ B ]
  private hfunc = | hh | ; h = _ ($) _ hfunc
  open Environment A using ( [ ] )
  open Environment B using () renaming ( [ ] to [ ]B )

  comm-hom-term : (t : Term X) (a : X → U[ A ]) → h ([ t ] ($) a) ≈ [ t ]B ($) (h o a)
  comm-hom-term (g x) a = refl
  comm-hom-term (node f t) a = goal
  where
    goal : h ([ node f t ] ($) a) ≈ ([ node f t ]B ($) (h o a))
    goal = begin
      h ([ node f t ] ($) a) ≈< compatible || hh || >
        (f ^ B)(λ i → h ([ t i ] ($) a)) ≈< cong (Interp B)(≡.refl , λ i → comm-hom-term (t i) a) >
        (f ^ B)(λ i → [ t i ]B ($) (h o a)) ≈< refl >
        ([ node f t ]B ($) (h o a)) ■

```

## 7 Model Theory and Equational Logic

(cf. the `Varieties.Func.SoundAndComplete` module of the `agda-algebras` library)

### 7.1 Basic definitions

Let  $S$  be a signature. By an *identity* or *equation* in  $S$  we mean an ordered pair of terms in a given context. For instance, if the context happens to be the type  $X : \text{Type } \chi$ , then an equation will be a pair of inhabitants of the domain of term algebra  $\mathbf{T} X$ .

We define an equation in Agda using the following record type with fields denoting the left-hand and right-hand sides of the equation, along with an equation “context” representing the underlying collection of variable symbols (cf. Andreas Abel’s formal proof of Birkhoff’s completeness theorem).

```

record Eq : Type (ov χ) where
  constructor _≈'_

```

```

field {cxt} : Type  $\chi$ 
lhs      : Term cxt
rhs      : Term cxt

```

```

open Eq public

```

We now define a type representing the notion of an equation  $p \approx \cdot q$  holding (when  $p$  and  $q$  are interpreted) in algebra  $\mathbf{A}$ .

If  $\mathbf{A}$  is an  $S$ -algebra we say that  $\mathbf{A}$  *satisfies*  $p \approx q$  provided for all environments  $\rho : X \rightarrow | \mathbf{A} |$  (assigning values in the domain of  $\mathbf{A}$  to variable symbols in  $X$ ) we have  $\llbracket p \rrbracket \langle \$ \rangle \rho \approx \llbracket q \rrbracket \langle \$ \rangle \rho$ . In this situation, we write  $\mathbf{A} \models (p \approx \cdot q)$  and say that  $\mathbf{A}$  *models* the identity  $p \approx q$ .

If  $\mathcal{K}$  is a class of algebras, all of the same signature, we write  $\mathcal{K} \models (p \approx \cdot q)$  if, for every  $\mathbf{A} \in \mathcal{K}$ , we have  $\mathbf{A} \models (p \approx \cdot q)$ .

Because a class of structures has a different type than a single structure, we must use a slightly different syntax to avoid overloading the relations  $\models$  and  $\approx$ . As a reasonable alternative to what we would normally express informally as  $\mathcal{K} \models p \approx q$ , we have settled on  $\mathcal{K} \models (p \approx \cdot q)$  to denote this relation. To reiterate, if  $\mathcal{K}$  is a class of  $S$ -algebras, we write  $\mathcal{K} \models (p \approx \cdot q)$  provided every  $\mathbf{A} \in \mathcal{K}$  satisfies  $\mathbf{A} \models (p \approx \cdot q)$ .

```

_⊨_ : (A : Algebra  $\alpha \rho^a$ ) (term-identity : Eq { $\chi$ }) → Type _
A ⊨ (p  $\approx \cdot$  q) = Equal p q where open Environment A

_⊨_ : Pred (Algebra  $\alpha \rho^a$ )  $\ell \rightarrow \text{Eq}\{\chi\} \rightarrow \text{Type } (\ell \sqcup \chi \sqcup \text{ov}(\alpha \sqcup \rho^a))$ 
 $\mathcal{K} \models \text{equ} = \forall \mathbf{A} \rightarrow \mathcal{K} \mathbf{A} \rightarrow \mathbf{A} \models \text{equ}$ 

```

We denote by  $\mathbf{A} \models \mathcal{E}$  the assertion that the algebra  $\mathbf{A}$  models every equation in a collection  $\mathcal{E}$  of equations.

```

_⊨_ : (A : Algebra  $\alpha \rho^a$ ) → { $\iota$  : Level} {I : Type  $\iota$ } → (I → Eq { $\chi$ }) → Type _
A ⊨  $\mathcal{E} = \forall i \rightarrow \text{Equal } (\text{lhs } (\mathcal{E} i)) (\text{rhs } (\mathcal{E} i)) \text{ where open Environment A}$ 

```

## 7.2 Equational theories and models

If  $\mathcal{K}$  denotes a class of structures, then  $\text{Th } \mathcal{K}$  represents the set of identities modeled by the members of  $\mathcal{K}$ .

```

Th : {X : Type  $\chi$ } → Pred (Algebra  $\alpha \rho^a$ )  $\ell \rightarrow \text{Pred}(\text{Term } X \times \text{Term } X) \_$ 
Th  $\mathcal{K} = \lambda (p, q) \rightarrow \mathcal{K} \models (p \approx \cdot q)$ 

Mod : {X : Type  $\chi$ } → Pred (Term  $X \times \text{Term } X$ )  $\ell \rightarrow \text{Pred}(\text{Algebra } \alpha \rho^a) \_$ 
Mod  $\mathcal{E} \mathbf{A} = \forall \{p, q\} \rightarrow (p, q) \in \mathcal{E} \rightarrow \text{Equal } p q \text{ where open Environment A}$ 

```

## 7.3 The entailment relation

Based on Andreas Abel's Agda formalization of Birkhoff's completeness theorem.

```

module _ { $\chi \iota$  : Level} where

data _⊢_⊢_⊢_ {I : Type  $\iota$ } ( $\mathcal{E} : I \rightarrow \text{Eq}$ ) : (X : Type  $\chi$ ) (p q : Term X) → Type ( $\iota \sqcup \text{ov } \chi$ ) where
  hyp :  $\forall i \rightarrow \text{let } p \approx \cdot q = \mathcal{E} i \text{ in } \mathcal{E} \vdash \_ \triangleright p \approx q$ 
  app :  $\forall \{ps qs\} \rightarrow (\forall i \rightarrow \mathcal{E} \vdash \Gamma \triangleright ps i \approx qs i) \rightarrow \mathcal{E} \vdash \Gamma \triangleright (\text{node } f \text{ ps}) \approx (\text{node } f \text{ qs})$ 
  sub :  $\forall \{p q\} \rightarrow \mathcal{E} \vdash \Delta \triangleright p \approx q \rightarrow \forall (\sigma : \text{Sub } \Gamma \Delta) \rightarrow \mathcal{E} \vdash \Gamma \triangleright (p [\sigma]) \approx (q [\sigma])$ 

```

```

 $\vdash_{\text{refl}} : \forall \{p\} \rightarrow \mathcal{E} \vdash \Gamma \triangleright p \approx p$ 
 $\vdash_{\text{sym}} : \forall \{p\ q : \text{Term } \Gamma\} \rightarrow \mathcal{E} \vdash \Gamma \triangleright p \approx q \rightarrow \mathcal{E} \vdash \Gamma \triangleright q \approx p$ 
 $\vdash_{\text{trans}} : \forall \{p\ q\ r : \text{Term } \Gamma\} \rightarrow \mathcal{E} \vdash \Gamma \triangleright p \approx q \rightarrow \mathcal{E} \vdash \Gamma \triangleright q \approx r \rightarrow \mathcal{E} \vdash \Gamma \triangleright p \approx r$ 

 $\vdash_{\approx} \text{IsEquiv} : \{X : \text{Type } \chi\} \{I : \text{Type } \iota\} \{\mathcal{E} : I \rightarrow \text{Eq}\} \rightarrow \text{IsEquivalence } (\mathcal{E} \vdash X \triangleright \_ \approx \_)$ 
 $\vdash_{\approx} \text{IsEquiv} = \text{record } \{ \text{refl} = \vdash_{\text{refl}} ; \text{sym} = \vdash_{\text{sym}} ; \text{trans} = \vdash_{\text{trans}} \}$ 

```

## 7.4 Soundness

In any model  $\mathbf{A}$  of the equations  $\mathcal{E}$  derived equality is actual equality (cf. Andreas Abel's Agda formalization of Birkhoff's completeness theorem).

```

module Soundness { $\chi$   $\alpha$   $\iota$  : Level} {I : Type  $\iota$ } ( $\mathcal{E} : I \rightarrow \text{Eq}\{\chi\}$ )
  ( $\mathbf{A} : \text{Algebra } \alpha \rho^a$ )  -- We assume an algebra  $\mathbf{A}$ 
  ( $\mathbf{V} : \mathbf{A} \models \mathcal{E}$ )        -- that models all equations in  $\mathcal{E}$ .
  where

  open SetoidReasoning  $\mathbb{D}[\mathbf{A}]$ 
  open Environment  $\mathbf{A}$ 
  open IsEquivalence using ( refl ; sym ; trans )

  sound :  $\forall \{p\ q\} \rightarrow \mathcal{E} \vdash \Gamma \triangleright p \approx q \rightarrow \mathbf{A} \models (p \approx \cdot q)$ 
  sound (hyp i) = V i
  sound (app es)  $\rho$  = cong (Interp  $\mathbf{A}$ ) ( $\equiv$ .refl ,  $\lambda i \rightarrow \text{sound } (\text{es } i) \rho$ )
  sound (sub {p = p}{q} Epq  $\sigma$ )  $\rho$  =
    begin
      [ p [  $\sigma$  ] ]  $\langle \$ \rangle$   $\rho \approx$  [ substitution p  $\sigma$   $\rho$  ]
      [ p ]  $\langle \$ \rangle$  [  $\sigma$  ]  $\rho \approx$  [ sound Epq ([  $\sigma$  ]  $\rho$ ) ]
      [ q ]  $\langle \$ \rangle$  [  $\sigma$  ]  $\rho \approx$  [ substitution q  $\sigma$   $\rho$  ]
      [ q [  $\sigma$  ] ]  $\langle \$ \rangle$   $\rho \blacksquare$ 
  sound (  $\vdash_{\text{refl}} \{p = p\}$  ) = refl EqualsEquiv {x = p}
  sound (  $\vdash_{\text{sym}} \{p = p\}\{q\}$  Epq ) = sym EqualsEquiv {x = p}\{q\} (sound Epq)
  sound (  $\vdash_{\text{trans}} \{p = p\}\{q\}\{r\}$  Epq Eqr ) = trans EqualsEquiv {i = p}\{q\}\{r\} (sound Epq)(sound Eqr)

```

## 7.5 The Closure Operators H, S, P and V

Fix a signature  $S$ , let  $\mathcal{K}$  be a class of  $S$ -algebras, and define

- $\mathbf{H} \mathcal{K}$  = algebras isomorphic to a homomorphic image of a member of  $\mathcal{K}$ ;
- $\mathbf{S} \mathcal{K}$  = algebras isomorphic to a subalgebra of a member of  $\mathcal{K}$ ;
- $\mathbf{P} \mathcal{K}$  = algebras isomorphic to a product of members of  $\mathcal{K}$ .

A straight-forward verification confirms that  $\mathbf{H}$ ,  $\mathbf{S}$ , and  $\mathbf{P}$  are *closure operators* (expansive, monotone, and idempotent). A class  $\mathcal{K}$  of  $S$ -algebras is said to be *closed under the taking of homomorphic images* provided  $\mathbf{H} \mathcal{K} \subseteq \mathcal{K}$ . Similarly,  $\mathcal{K}$  is *closed under the taking of subalgebras* (resp., *arbitrary products*) provided  $\mathbf{S} \mathcal{K} \subseteq \mathcal{K}$  (resp.,  $\mathbf{P} \mathcal{K} \subseteq \mathcal{K}$ ). The operators  $\mathbf{H}$ ,  $\mathbf{S}$ , and  $\mathbf{P}$  can be composed with one another repeatedly, forming yet more closure operators.

An algebra is a homomorphic image (resp., subalgebra; resp., product) of every algebra to which it is isomorphic. Thus, the class  $\mathbf{H} \mathcal{K}$  (resp.,  $\mathbf{S} \mathcal{K}$ ; resp.,  $\mathbf{P} \mathcal{K}$ ) is closed under isomorphism.

A *variety* is a class of  $S$ -algebras that is closed under the taking of homomorphic images, subalgebras, and arbitrary products. To represent varieties we define types for the closure

operators **H**, **S**, and **P** that are composable. Separately, we define a type **V** which represents closure under all three operators, **H**, **S**, and **P**. Thus, if  $\mathcal{K}$  is a class of  $S$ -algebras<sup>4</sup>, then  $\mathbf{V} \mathcal{K} := \mathbf{H} (\mathbf{S} (\mathbf{P} \mathcal{K}))$ , and  $\mathcal{K}$  is a variety iff  $\mathbf{V} \mathcal{K} \subseteq \mathcal{K}$ .

We now define the type **H** to represent classes of algebras that include all homomorphic images of algebras in the class—i.e., classes that are closed under the taking of homomorphic images—the type **S** to represent classes of algebras that closed under the taking of subalgebras, and the type **P** to represent classes of algebras closed under the taking of arbitrary products.

```

module _ {α ρa β ρb : Level} where
  private a = α ⊔ ρa ; b = β ⊔ ρb

  H : ∀ ℓ → Pred(Algebra α ρa) (a ⊔ ov ℓ) → Pred(Algebra β ρb) (b ⊔ ov(a ⊔ ℓ))
  H _ ℓ B = Σ[ A ∈ Algebra α ρa ] A ∈ ℓ × B IsHomImageOf A

  S : ∀ ℓ → Pred(Algebra α ρa) (a ⊔ ov ℓ) → Pred(Algebra β ρb) (b ⊔ ov(a ⊔ ℓ))
  S _ ℓ B = Σ[ A ∈ Algebra α ρa ] A ∈ ℓ × B ≤ A

  P : ∀ ℓ ι → Pred(Algebra α ρa) (a ⊔ ov ℓ) → Pred(Algebra β ρb) (b ⊔ ov(a ⊔ ℓ ⊔ ι))
  P _ ι ℓ B = Σ[ I ∈ Type ι ] (Σ[ sI ∈ (I → Algebra α ρa) ] (∀ i → sI i ∈ ℓ) × (B ≅ ∏ sI))

module _ {α ρa β ρb γ ρc δ ρd : Level} where
  private a = α ⊔ ρa ; b = β ⊔ ρb ; c = γ ⊔ ρc ; d = δ ⊔ ρd

  V : ∀ ℓ ι → Pred(Algebra α ρa) (a ⊔ ov ℓ) → Pred(Algebra δ ρd) (d ⊔ ov(a ⊔ b ⊔ c ⊔ ℓ ⊔ ι))
  V ℓ ι ℓ = H{γ}{ρc}{δ}{ρd} (a ⊔ b ⊔ ℓ ⊔ ι) (S{β}{ρb} (a ⊔ ℓ ⊔ ι) (P ℓ ι ℓ))

```

### 7.5.1 Idempotence of S

**S** is a closure operator. The facts that **S** is monotone and expansive won't be needed, so we omit the proof of these facts. However, we will make use of idempotence of **S**, so we prove that property as follows.

```

S-idem : {ℓ : Pred (Algebra α ρa) (α ⊔ ρa ⊔ ov ℓ)}
  → S{β = γ}{ρc} (α ⊔ ρa ⊔ ℓ) (S{β = β}{ρb} ℓ ℓ) ⊆ S{β = γ}{ρc} ℓ ℓ

S-idem (A , (B , sB , A ≤ B) , x ≤ A) = B , (sB , ≤-trans x ≤ A A ≤ B)

```

### 7.5.2 Algebraic invariance of $\models$

The binary relation  $\models$  would be practically useless if it were not an *algebraic invariant* (i.e., invariant under isomorphism). Let us now verify that the models relation we defined above has this essential property.

```

module _ {X : Type χ} {A : Algebra α ρa} {B : Algebra β ρb} (p q : Term X) where

  ≡-l-invar : A ⊨ (p ≈ · q) → A ≅ B → B ⊨ (p ≈ · q)
  ≡-l-invar Apq (mkiso fh gh f~g g~f) ρ =
  begin
    [ p ]2 ($) ρ ≈< cong [ p ]2 (λ x → f~g (ρ x)) >
    [ p ]2 ($) (f ∘ (g ∘ ρ)) ≈< comm-hom-term fh p (g ∘ ρ) >
    f ([ p ]1 ($) (g ∘ ρ)) ≈< cong | fh | (Apq (g ∘ ρ)) >
    f ([ q ]1 ($) (g ∘ ρ)) ≈< comm-hom-term fh q (g ∘ ρ) >
    [ q ]2 ($) (f ∘ (g ∘ ρ)) ≈< cong [ q ]2 (λ x → f~g (ρ x)) >
    [ q ]2 ($) ρ ■

```

```

where
open Environment A using () renaming ( [ ] to [ ]1 )
open Environment B using () renaming ( [ ] to [ ]2 )
open SetoidReasoning D[ B ]
private f = _⟨$⟩_ | fh | ; g = _⟨$⟩_ | gh |

```

### 7.5.3 Subalgebraic invariance of $\models$

Identities modeled by an algebra  $\mathbf{A}$  are also modeled by every subalgebra of  $\mathbf{A}$ , which fact can be formalized as follows.

```

module _ {X : Type χ} {A : Algebra α ρa} {B : Algebra β ρb} {p q : Term X} where
  |-S-invar : A ⊨ (p ≈ · q) → B ≤ A → B ⊨ (p ≈ · q)
  |-S-invar Apq B≤A b = goal
  where
    open Setoid D[ A ] using ( _≈_ )
    open Environment A using () renaming ( [ ] to [ ]a )
    open Setoid D[ B ] using () renaming ( _≈_ to _≈b_ )
    open Environment B using ( [ ] )
    open SetoidReasoning D[ A ]
    hh : hom B A
    hh = | B≤A |
    h = _⟨$⟩_ | hh |
    ξ : ∀ b → h ( [ p ] ⟨$⟩ b ) ≈ h ( [ q ] ⟨$⟩ b )
    ξ b = begin
      h ( [ p ] ⟨$⟩ b )      ≈⟨ comm-hom-term hh p b ⟩
      [ p ]a ⟨$⟩ (h ○ b)    ≈⟨ Apq (h ○ b) ⟩
      [ q ]a ⟨$⟩ (h ○ b)    ≈✓⟨ comm-hom-term hh q b ⟩
      h ( [ q ] ⟨$⟩ b )      ■
    goal : [ p ] ⟨$⟩ b ≈b [ q ] ⟨$⟩ b
    goal = || B≤A || (ξ b)

```

### 7.5.4 Product invariance of $\models$

An identity satisfied by all algebras in an indexed collection is also satisfied by the product of algebras in that collection.

```

module _ {X : Type χ} {I : Type ℓ} {A : I → Algebra α ρa} {p q : Term X} where
  |-P-invar : (∀ i → A i ⊨ (p ≈ · q)) → ∏ A ⊨ (p ≈ · q)
  |-P-invar A pq a = goal
  where
    open Environment (∏ A) using () renaming ( [ ] to [ ]1 )
    open Environment using ( [ ] )
    open Setoid D[ ∏ A ] using ( _≈_ )
    open SetoidReasoning D[ ∏ A ]
    ξ : (λ i → ( [ A i ] p ) ⟨$⟩ (λ x → (a x) i)) ≈ (λ i → ( [ A i ] q ) ⟨$⟩ (λ x → (a x) i))
    ξ = λ i → A pq i (λ x → (a x) i)
    goal : [ p ]1 ⟨$⟩ a ≈ [ q ]1 ⟨$⟩ a
    goal = begin
      [ p ]1 ⟨$⟩ a      ≈⟨ interp-prod A p a ⟩

```

$$\begin{aligned}
& (\lambda i \rightarrow (\llbracket \mathcal{A} i \rrbracket p) \langle \$ \rangle (\lambda x \rightarrow (a x) i)) \approx \langle \xi \rangle \\
& (\lambda i \rightarrow (\llbracket \mathcal{A} i \rrbracket q) \langle \$ \rangle (\lambda x \rightarrow (a x) i)) \approx \langle \text{interp-prod } \mathcal{A} q a \rangle \\
& \llbracket q \rrbracket_1 \langle \$ \rangle a \quad \blacksquare
\end{aligned}$$

### 7.5.5 PS ⊆ SP

Another important fact we will need about the operators **S** and **P** is that a product of subalgebras of algebras in a class  $\mathcal{K}$  is a subalgebra of a product of algebras in  $\mathcal{K}$ . We denote this inclusion by **PS**⊆**SP**, which we state and prove as follows.

```

module _ {K : Pred(Algebra α ρa) (α ⊔ ρa ⊔ ov ℓ)} where
private
  a = α ⊔ ρa
  oal = ov (a ⊔ ℓ)

PS⊆SP : P (a ⊔ ℓ) oal (S{β = α}{ρa} ℓ K) ⊆ S oal (P ℓ oal K)
PS⊆SP {B} (I , (A , sA , B ≅ ⊔ A)) = Goal
where
  B : I → Algebra α ρa
  B i = | sA i |
  kB : (i : I) → B i ∈ K
  kB i = fst || sA i ||
  ⊔ A ≤ ⊔ B : ⊔ A ≤ ⊔ B
  ⊔ A ≤ ⊔ B = ⊔-≤ λ i → snd || sA i ||
  Goal : B ∈ S{β = oal}{oal} oal (P {β = oal}{oal} ℓ oal K)
  Goal = ⊔ B , (I , (B , (kB , ≅-refl))) , (≅-trans-≤ B ≅ ⊔ A ⊔ A ≤ ⊔ B)

```

### 7.5.6 Identity preservation

The classes  $\mathbf{H} \mathcal{K}$ ,  $\mathbf{S} \mathcal{K}$ ,  $\mathbf{P} \mathcal{K}$ , and  $\mathbf{V} \mathcal{K}$  all satisfy the same set of equations. We will only use a subset of the inclusions used to prove this fact. (For a complete proof, see the **Varieties.Func.Preservation** module of the *agda-algebras* library.)

#### 7.5.6.1 H preserves identities

First we prove that the closure operator **H** is compatible with identities that hold in the given class.

```

module _ {X : Type X} {K : Pred(Algebra α ρa) (α ⊔ ρa ⊔ ov ℓ)} {p q : Term X} where

H-id1 : K || (p ≈ · q) → (H {β = α}{ρa} ℓ K) || (p ≈ · q)
H-id1 σ B (A , kA , BimgOfA) ρ = B||pq
where
  IH : A || (p ≈ · q)
  IH = σ A kA
  open Environment A using () renaming ( ⊔ to ⊔1 )
  open Environment B using ( ⊔ )
  open Setoid D[ B ] using ( ≈ )
  open SetoidReasoning D[ B ]

φ : hom A B
φ = | BimgOfA |

```

```

 $\varphi E : \text{IsSurjective} \mid \varphi \mid$ 
 $\varphi E = \parallel \text{BimgOfA} \parallel$ 
 $\varphi^{-1} : \mathbb{U}[\mathbf{B}] \rightarrow \mathbb{U}[\mathbf{A}]$ 
 $\varphi^{-1} = \text{SurjInv} \mid \varphi \mid \varphi E$ 

 $\zeta : \forall x \rightarrow ( \mid \varphi \mid \langle \$ \rangle (\varphi^{-1} \circ \rho) x ) \approx \rho x$ 
 $\zeta = \lambda \_ \rightarrow \text{InvsInverse}^r \varphi E$ 

 $B \models pq : ( \parallel p \parallel \langle \$ \rangle \rho ) \approx ( \parallel q \parallel \langle \$ \rangle \rho )$ 
 $B \models pq = \text{begin}$ 
   $\parallel p \parallel \langle \$ \rangle \rho \approx \langle \text{cong} \parallel p \parallel \zeta$ 
   $\parallel p \parallel \langle \$ \rangle (\lambda x \rightarrow ( \mid \varphi \mid \langle \$ \rangle (\varphi^{-1} \circ \rho) x )) \approx \langle \text{comm-hom-term} \varphi p (\varphi^{-1} \circ \rho)$ 
   $\mid \varphi \mid \langle \$ \rangle ( \parallel p \parallel_1 \langle \$ \rangle (\varphi^{-1} \circ \rho) ) \approx \langle \text{cong} \mid \varphi \mid ( \text{IH} (\varphi^{-1} \circ \rho) )$ 
   $\mid \varphi \mid \langle \$ \rangle ( \parallel q \parallel_1 \langle \$ \rangle (\varphi^{-1} \circ \rho) ) \approx \langle \text{comm-hom-term} \varphi q (\varphi^{-1} \circ \rho)$ 
   $\parallel q \parallel \langle \$ \rangle (\lambda x \rightarrow ( \mid \varphi \mid \langle \$ \rangle (\varphi^{-1} \circ \rho) x )) \approx \langle \text{cong} \parallel q \parallel \zeta$ 
   $\parallel q \parallel \langle \$ \rangle \rho$ 

```

### 7.5.6.2 S preserves identities

$\text{S-id1} : \mathcal{K} \models (p \approx \cdot q) \rightarrow (\text{S } \{\beta = \alpha\} \{\rho^a\} \ell \mathcal{K}) \models (p \approx \cdot q)$   
 $\text{S-id1 } \sigma \mathbf{A} (\mathbf{A}, \mathbf{kA}, \mathbf{B} \leq \mathbf{A}) = \models \text{S-invar} \{p = p\} \{q\} (\sigma \mathbf{A} \mathbf{kA}) \mathbf{B} \leq \mathbf{A}$

The obvious converse is barely worth the bits needed to formalize it, but we will use it below, so let's prove it now.

$\text{S-id2} : \text{S } \ell \mathcal{K} \models (p \approx \cdot q) \rightarrow \mathcal{K} \models (p \approx \cdot q)$   
 $\text{S-id2 } \text{Spq } \mathbf{A} \mathbf{kA} = \text{Spq } \mathbf{A} (\mathbf{A}, (\mathbf{kA}, \leq\text{-reflexive}))$

### 7.5.6.3 P preserves identities

$\text{P-id1} : \forall \{\iota\} \rightarrow \mathcal{K} \models (p \approx \cdot q) \rightarrow \text{P } \{\beta = \alpha\} \{\rho^a\} \ell \iota \mathcal{K} \models (p \approx \cdot q)$   
 $\text{P-id1 } \sigma \mathbf{A} (\mathbf{I}, \mathcal{A}, \mathbf{kA}, \mathbf{A} \cong \sqcap \mathbf{A}) = \models \text{I-invar } \mathbf{A} p q \text{ IH } (\cong\text{-sym } \mathbf{A} \cong \sqcap \mathbf{A})$   
 where  
 $\text{ih} : \forall i \rightarrow \mathcal{A} i \models (p \approx \cdot q)$   
 $\text{ih } i = \sigma (\mathcal{A} i) (\mathbf{kA} i)$   
 $\text{IH} : \sqcap \mathcal{A} \models (p \approx \cdot q)$   
 $\text{IH} = \models \text{P-invar } \mathcal{A} \{p\} \{q\} \text{ ih}$

### 7.5.6.4 V preserves identities

Finally, we prove the analogous preservation lemmas for the closure operator V.

```

module _ {X : Type} {χ} {ι : Level} {K : Pred (Algebra α ρa) (α ⊔ ρa ⊔ ov ℓ)} {p q : Term X} where
  private
    aℓι = α ⊔ ρa ⊔ ℓ ⊔ ι

  V-id1 : K ⊨ (p ≈ · q) → V ℓ ι K ⊨ (p ≈ · q)
  V-id1 σ B (A, (⊔A, p⊔A, A ≤ ⊔A), BimgA) =
    H-id1 {ℓ = aℓι} {K = S aℓι (P {β = α} {ρa} ℓ ι K)} {p = p} {q} spK ⊨ pq B (A, (spA, BimgA))
  where
    spA : A ∈ S aℓι (P {β = α} {ρa} ℓ ι K)

```

```

spA =  $\prod A$  , (p  $\prod$  A , A  $\leq$   $\prod$  A)
spK  $\models$  pq : S ali (P  $\ell$   $\iota$  K)  $\models$  (p  $\approx$   $\cdot$  q)
spK  $\models$  pq = S-id1 { $\ell$  = ali} {p = p} {q} (P-id1 { $\ell$  =  $\ell$ } {K = K} {p = p} {q}  $\sigma$ )

```

### 7.5.7 Th $\mathcal{K} \subseteq \text{Th}(\mathbf{V} \mathcal{K})$

From V-id1 it follows that if  $\mathcal{K}$  is a class of algebras, then the set of identities modeled by the algebras in  $\mathcal{K}$  is contained in the set of identities modeled by the algebras in  $\mathbf{V} \mathcal{K}$ . In other terms,  $\text{Th} \mathcal{K} \subseteq \text{Th}(\mathbf{V} \mathcal{K})$ . We formalize this observation as follows.

```

classIds- $\subseteq$ -Vlds : K  $\models$  (p  $\approx$   $\cdot$  q)  $\rightarrow$  (p , q)  $\in$  Th (V  $\ell$   $\iota$  K)
classIds- $\subseteq$ -Vlds pKq A = V-id1 pKq A

```

## 8 Free Algebras

### 8.1 The absolutely free algebra $\mathbf{T} X$

The term algebra  $\mathbf{T} X$  is *absolutely free* (or *universal*, or *initial*) for algebras in the signature  $S$ . That is, for every  $S$ -algebra  $A$ , the following hold.

- Every function from  $X$  to  $|A|$  lifts to a homomorphism from  $\mathbf{T} X$  to  $A$ .
- The homomorphism that exists by item 1 is unique.

We now prove this in Agda, starting with the fact that every map from  $X$  to  $|A|$  lifts to a map from  $|\mathbf{T} X|$  to  $|A|$  in a natural way, by induction on the structure of the given term.

```

module _ {X : Type} {A : Algebra  $\alpha$   $\rho^a$ } (h : X  $\rightarrow$  U[ A ]) where
  open Setoid  $\mathbb{D}[ A ]$  using (  $\_ \approx \_$  ; reflexive ; refl ; trans )

  free-lift : U[  $\mathbf{T} X$  ]  $\rightarrow$  U[ A ]
  free-lift (g x) = h x
  free-lift (node f t) = (f ^ A) ( $\lambda$  i  $\rightarrow$  free-lift (t i))

  free-lift-func :  $\mathbb{D}[ \mathbf{T} X ] \rightarrow \mathbb{D}[ A ]$ 
  free-lift-func ($) x = free-lift x
  cong free-lift-func = flcong
  where
    flcong :  $\forall \{s t\} \rightarrow s \doteq t \rightarrow$  free-lift s  $\approx$  free-lift t
    flcong ( $\_ \doteq \_$ .rfl x) = reflexive ( $\equiv$ .cong h x)
    flcong ( $\_ \doteq \_$ .gnl x) = cong (Interp A) ( $\equiv$ .refl , ( $\lambda$  i  $\rightarrow$  flcong (x i)))

```

Naturally, at the base step of the induction, when the term has the form  $g x$ , the free lift of  $h$  agrees with  $h$ . For the inductive step, when the given term has the form  $node f t$ , the free lift is defined as follows: Assuming (the induction hypothesis) that we know the image of each subterm  $t i$  under the free lift of  $h$ , define the free lift at the full term by applying  $f ^ A$  to the images of the subterms.

The free lift so defined is a homomorphism by construction. Indeed, here is the trivial proof.

```

lift-hom : hom ( $\mathbf{T} X$ ) A
lift-hom = free-lift-func , hhom
where

```



```

hfunc :  $\mathbb{D}[\mathbf{T} X] \rightarrow \mathbb{D}[\mathbf{A}]$ 
hfunc = free-lift-func

hcomp : compatible-map ( $\mathbf{T} X$ )  $\mathbf{A}$  free-lift-func
hcomp {f}{a} = cong (Interp  $\mathbf{A}$ ) ( $\equiv$ .refl ,  $(\lambda i \rightarrow (\text{cong free-lift-func})\{a\} i) \dot{=} \text{isRefl}$ ))

hhom : IsHom ( $\mathbf{T} X$ )  $\mathbf{A}$  hfunc
hhom = mkhom ( $\lambda\{f\}\{a\} \rightarrow \text{hcomp}\{f\}\{a\}$ )

module _ {X : Type  $\chi$ } {A : Algebra  $\alpha \rho^a$ } where
  open Setoid  $\mathbb{D}[\mathbf{A}]$  using (  $\approx$  ; refl )
  open Environment  $\mathbf{A}$  using (  $\llbracket \_ \rrbracket$  )

  free-lift-interp : ( $\eta : X \rightarrow \mathbb{U}[\mathbf{A}]$ )(p : Term X)  $\rightarrow \llbracket p \rrbracket \langle \$ \rangle \eta \approx (\text{free-lift } \{A = A\} \eta) p$ 
  free-lift-interp  $\eta$  ( $g\ x$ ) = refl
  free-lift-interp  $\eta$  (node f t) = cong (Interp  $\mathbf{A}$ ) ( $\equiv$ .refl , (free-lift-interp  $\eta$ )  $\circ$  t)

```

## 8.2 The relatively free algebra $\mathbb{F}[X]$

We now define the algebra  $\mathbb{F}[X]$ , which represents the relatively free algebra. Here, as above,  $X$  plays the role of an arbitrary nonempty collection of variables. (It would suffice to take  $X$  to be the cardinality of the largest algebra in  $\mathcal{K}$ , but since we don't know that cardinality, we leave  $X$  arbitrary for now.)

```

module FreeAlgebra { $\chi$  : Level}{ $\iota$  : Level}{I : Type  $\iota$ }( $\mathcal{E} : I \rightarrow \text{Eq}$ ) where
  open Algebra

  FreeDomain : Type  $\chi \rightarrow \text{Setoid } \_ \_$ 
  FreeDomain X = record { Carrier = Term X
                        ;  $\approx$  =  $\mathcal{E} \vdash X \triangleright \approx$ 
                        ; isEquivalence =  $\vdash \triangleright \approx \text{IsEquiv}$  }

```

The interpretation of an operation is simply the operation itself. This works since  $\mathcal{E} \vdash X$   $\triangleright \approx$  is a congruence.

```

FreeInterp :  $\forall \{X\} \rightarrow \langle S \rangle (\text{FreeDomain } X) \rightarrow \text{FreeDomain } X$ 
FreeInterp  $\langle \$ \rangle$  (f , ts) = node f ts
cong FreeInterp ( $\equiv$ .refl , h) = app h

 $\mathbb{F}[\_]$  : Type  $\chi \rightarrow \text{Algebra } (\text{ov } \chi) (\iota \sqcup \text{ov } \chi)$ 
Domain  $\mathbb{F}[X] = \text{FreeDomain } X$ 
Interp  $\mathbb{F}[X] = \text{FreeInterp}$ 

```

## 8.3 Basic properties of free algebras

```

module FreeHom ( $\chi$  : Level) { $\mathcal{K} : \text{Pred}(\text{Algebra } \alpha \rho^a) (\alpha \sqcup \rho^a \sqcup \text{ov } \ell)$ } where
  private  $\iota = \text{ov}(\chi \sqcup \alpha \sqcup \rho^a \sqcup \ell)$ 
  open Eq

   $\mathcal{J} : \text{Type } \iota$  – indexes the collection of equations modeled by  $\mathcal{K}$ 
   $\mathcal{J} = \Sigma[\text{eq} \in \text{Eq}\{\chi\}] \mathcal{K} \models ((\text{lhs eq}) \approx \cdot (\text{rhs eq}))$ 

   $\mathcal{E} : \mathcal{J} \rightarrow \text{Eq}$ 
   $\mathcal{E} (\text{eqv} , p) = \text{eqv}$ 

```

```

 $\mathcal{E} \vdash \_ \triangleright \text{Th} \mathcal{K} : (X : \text{Type } \chi) \rightarrow \forall \{p\ q\} \rightarrow \mathcal{E} \vdash X \triangleright p \approx q \rightarrow \mathcal{K} \models (p \approx \cdot q)$ 
 $\mathcal{E} \vdash [X] \triangleright \text{Th} \mathcal{K} \times \mathbf{A} \text{ kA} = \text{sound } (\lambda i \ \rho \rightarrow \parallel i \parallel \mathbf{A} \text{ kA } \rho) \times$ 
  where open Soundness  $\mathcal{E} \ \mathbf{A}$ 
open FreeAlgebra  $\{\iota = \iota\} \{l = \mathcal{J}\} \ \mathcal{E} \text{ using } (\mathbb{F} \_)$ 

```

### 8.3.1 The natural epimorphism from $\mathbf{T} \ X$ to $\mathbb{F}[X]$

We now define the natural epimorphism from  $\mathbf{T} \ X$  onto the relatively free algebra  $\mathbb{F}[X]$  and prove that the kernel of this morphism is the congruence of  $\mathbf{T} \ X$  defined by the identities modeled by  $(S \ \mathcal{K})$ , hence by  $\mathcal{K}$ .

```

epiF $\_$  : (X : Type  $\chi$ )  $\rightarrow$  epi (T X)  $\mathbb{F}[X]$ 
epiF $[X]$  = h , hepi
  where
    open Algebra (T X) using () renaming (Domain to TX)
    open Setoid TX using () renaming (  $\_ \approx \_$  to  $\_ \approx_0 \_$  ; refl to refl $_0$  )
    open Algebra  $\mathbb{F}[X]$  using () renaming ( Domain to F )
    open Setoid F using () renaming (  $\_ \approx \_$  to  $\_ \approx_1 \_$  ; refl to refl $_1$  )
    open  $\_ \doteq \_$ 

    c :  $\forall \{x\ y\} \rightarrow x \approx_0 y \rightarrow x \approx_1 y$ 
    c (rfl {x}{y}  $\equiv$  .refl) = refl $_1$ 
    c (gnl {f}{s}{t} x) = cong (Interp  $\mathbb{F}[X]$ ) ( $\equiv$  .refl , c  $\circ$  x)

    h : TX  $\rightarrow$  F
    h = record { f = id ; cong = c }

    hepi : IsEpi (T X)  $\mathbb{F}[X]$  h
    compatible (isHom hepi) = cong h refl $_0$ 
    isSurjective hepi {y} = eq y refl $_1$ 

    homF $\_$  : (X : Type  $\chi$ )  $\rightarrow$  hom (T X)  $\mathbb{F}[X]$ 
    homF $[X]$  = IsEpi.HomReduct  $\parallel$  epiF $[X]$   $\parallel$ 

    homF $\_$ -is-epic : (X : Type  $\chi$ )  $\rightarrow$  IsSurjective  $\mid$  homF $[X]$   $\mid$ 
    homF $[X]$ -is-epic = IsEpi.isSurjective (snd (epiF $[X]$ ))

```

As promised, we prove that the kernel of the natural epimorphism is the congruence defined by the identities modelled by  $\mathcal{K}$ .

```

class-models-kernel :  $\forall \{X\ p\ q\} \rightarrow (p , q) \in \text{ker} \mid \text{homF}[X] \mid \rightarrow \mathcal{K} \models (p \approx \cdot q)$ 
class-models-kernel {X = X} {p}{q} pKq =  $\mathcal{E} \vdash [X] \triangleright \text{Th} \mathcal{K} \text{ pKq}$ 

kernel-in-theory :  $\{X : \text{Type } \chi\} \rightarrow \text{ker} \mid \text{homF}[X] \mid \subseteq \text{Th } (\bigvee \ell \ \iota \ \mathcal{K})$ 
kernel-in-theory {X = X} {p , q} pKq vkA x = classIds  $\subseteq$  VIds  $\{\ell = \ell\} \{p = p\} \{q\}$ 
  (class-models-kernel pKq) vkA x

module  $\_$  {X : Type  $\chi$ } {A : Algebra  $\alpha \ \rho^a$ } {sA : A  $\in S \ \{\beta = \alpha\} \{\rho^a\} \ell \ \mathcal{K}$ } where
  open Environment A using (Equal)
  kerF $\subseteq$ Equal :  $\forall \{p\ q\} \rightarrow (p , q) \in \text{ker} \mid \text{homF}[X] \mid \rightarrow \text{Equal } p\ q$ 
  kerF $\subseteq$ Equal {p = p} {q} x = S-id1  $\{\ell = \ell\} \{p = p\} \{q\}$  ( $\mathcal{E} \vdash [X] \triangleright \text{Th} \mathcal{K} \ x$ ) A sA

 $\mathcal{K} \models \rightarrow \mathcal{E} \vdash$  :  $\{X : \text{Type } \chi\} \rightarrow \forall \{p\ q\} \rightarrow \mathcal{K} \models (p \approx \cdot q) \rightarrow \mathcal{E} \vdash X \triangleright p \approx q$ 
 $\mathcal{K} \models \rightarrow \mathcal{E} \vdash$  {p = p} {q} pKq = hyp ((p  $\approx \cdot$  q) , pKq) where open  $\_ \vdash \_ \triangleright \_ \approx \_$  using (hyp)

```

## 8.3.2 The universal property

```

module _ { A : Algebra (α ⊔ ρa ⊔ ℓ) (α ⊔ ρa ⊔ ℓ) } { K : Pred (Algebra α ρa) (α ⊔ ρa ⊔ ov ℓ) } where
  private ι = ov (α ⊔ ρa ⊔ ℓ)

  open FreeHom {ℓ = ℓ} (α ⊔ ρa ⊔ ℓ) { K }
  open FreeAlgebra {ι = ι} {ℓ = ℓ} ℱ using ( ℱ[_] )
  open Setoid ℱ[ A ] using ( trans ; sym ; refl ) renaming ( Carrier to A )

ℱ-ModTh-epi : A ∈ Mod (Th (V ℓ ι K))
  → epi ℱ[ A ] A
ℱ-ModTh-epi A ∈ ModThK = φ , isEpi
where
  φ : ℱ[ A ] → ℱ[ A ]
  _($)_ φ = free-lift { A = A } id
  cong φ {p} {q} pq = trans ( sym (free-lift-interp { A = A } id p) )
    ( trans ( A ∈ ModThK {p = p} {q} (kernel-in-theory pq) id )
      ( free-lift-interp { A = A } id q ) )
  isEpi : IsEpi ℱ[ A ] A φ
  compatible (isHom isEpi) = cong (Interp A) (≡.refl , (λ _ → refl))
  isSurjective isEpi {y} = eq (g y) refl

ℱ-ModTh-epi-lift : A ∈ Mod (Th (V ℓ ι K)) → epi ℱ[ A ] (Lift-Alg A ι ι)
ℱ-ModTh-epi-lift A ∈ ModThK = o-epi (ℱ-ModTh-epi (λ {p q} → A ∈ ModThK {p = p} {q})) ToLift-epi

```

## 8.4 Products of classes of algebras

We want to pair each  $(\mathbf{A}, p)$  (where  $p : \mathbf{A} \in \mathcal{S} \mathcal{K}$ ) with an environment  $\rho : X \rightarrow |\mathbf{A}|$  so that we can quantify over all algebras *and* all assignments of values in the domain  $|\mathbf{A}|$  to variables in  $X$ .

```

module _ (K : Pred (Algebra α ρa) (α ⊔ ρa ⊔ ov ℓ)) { X : Type (α ⊔ ρa ⊔ ℓ) } where
  private ι = ov (α ⊔ ρa ⊔ ℓ)
  open FreeHom {ℓ = ℓ} (α ⊔ ρa ⊔ ℓ) { K }
  open FreeAlgebra {ι = ι} {ℓ = ℓ} ℱ using ( ℱ[_] )
  open Environment using ( Env )

  ℱ+ : Type ι
  ℱ+ = Σ[ A ∈ (Algebra α ρa) ] ( A ∈ S ℓ K ) × ( Carrier ( Env A X ) )

  ℱ+ : ℱ+ → Algebra α ρa
  ℱ+ i = | i |

  ℱ : Algebra ι ι
  ℱ = ∏ ℱ+

```

Next we define a useful type, **skEqual**, which we use to represent a term identity  $p \approx q$  for any given  $i = (\mathbf{A}, s\mathbf{A}, \rho)$  (where  $\mathbf{A}$  is an algebra,  $s\mathbf{A} : \mathbf{A} \in \mathcal{S} \mathcal{K}$  is a proof that  $\mathbf{A}$  belongs to  $\mathcal{S} \mathcal{K}$ , and  $\rho$  is a mapping from  $X$  to the domain of  $\mathbf{A}$ ). Then we prove **AllEqual**  $\subseteq \ker \mathbb{F}$  which asserts that if the identity  $p \approx q$  holds in all  $\mathbf{A} \in \mathcal{S} \mathcal{K}$  (for all environments), then  $p \approx q$  holds in the relatively free algebra  $\mathbb{F}[X]$ ; equivalently, the pair  $(p, q)$  belongs to the kernel of the natural homomorphism from  $\mathbf{T} X$  onto  $\mathbb{F}[X]$ . We will use this fact below to prove

that there is a monomorphism from  $\mathbb{F}[X]$  into  $\mathfrak{C}$ , and thus  $\mathbb{F}[X]$  is a subalgebra of  $\mathfrak{C}$ , so belongs to  $S(P\mathcal{K})$ .

```

skEqual : (i :  $\mathfrak{I}^+$ )  $\rightarrow \forall \{p\ q\} \rightarrow \text{Type } \rho^a$ 
skEqual i {p}{q} =  $\llbracket p \rrbracket \langle \$ \rangle \text{snd } \llbracket i \rrbracket \approx \llbracket q \rrbracket \langle \$ \rangle \text{snd } \llbracket i \rrbracket$ 
  where
    open Setoid  $\mathbb{D}[\mathfrak{A}^+ i]$  using (  $\_ \approx \_$  )
    open Environment ( $\mathfrak{A}^+ i$ ) using (  $\llbracket \_ \rrbracket$  )

AllEqual $\subseteq$ kerF :  $\forall \{p\ q\} \rightarrow (\forall i \rightarrow \text{skEqual } i \{p\}\{q\}) \rightarrow (p, q) \in \text{ker} \mid \text{homF}[X] \mid$ 
AllEqual $\subseteq$ kerF {p}{q} x = Goal
  where
    open Setoid  $\mathbb{D}[\mathbb{F}[X]]$  using (  $\_ \approx \_$  )
     $S\mathcal{K} \models_{pq} : S\{\beta = \alpha\}\{\rho^a\} \ell \mathcal{K} \models (p \approx \cdot q)$ 
     $S\mathcal{K} \models_{pq} \mathbf{A} \text{ sA } \rho = x (\mathbf{A}, \text{sA}, \rho)$ 
    Goal :  $p \approx q$ 
    Goal =  $\mathcal{K} \models \rightarrow \mathcal{K} \vdash (S\text{-id2}\{\ell = \ell\}\{p = p\}\{q\} S\mathcal{K} \models_{pq})$ 

hom $\mathfrak{C}$  :  $\text{hom}(\mathbf{T} X) \mathfrak{C}$ 
hom $\mathfrak{C}$  =  $\sqcap\text{-hom-co } \mathfrak{A}^+ h$ 
  where
    h :  $\forall i \rightarrow \text{hom}(\mathbf{T} X) (\mathfrak{A}^+ i)$ 
    h i = lift-hom (snd  $\llbracket i \rrbracket$ )

kerF $\subseteq$ ker $\mathfrak{C}$  :  $\text{ker} \mid \text{homF}[X] \mid \subseteq \text{ker} \mid \text{hom}\mathfrak{C} \mid$ 
kerF $\subseteq$ ker $\mathfrak{C}$  {p, q} pKq ( $\mathbf{A}, \text{sA}, \rho$ ) = Goal
  where
    open Setoid  $\mathbb{D}[\mathbf{A}]$  using (  $\_ \approx \_ ; \text{sym} ; \text{trans}$  )
    open Environment  $\mathbf{A}$  using (  $\llbracket \_ \rrbracket$  )
    fl :  $\forall t \rightarrow \llbracket t \rrbracket \langle \$ \rangle \rho \approx \text{free-lift } \rho t$ 
    fl t = free-lift-interp { $\mathbf{A} = \mathbf{A}$ }  $\rho t$ 
    subgoal :  $\llbracket p \rrbracket \langle \$ \rangle \rho \approx \llbracket q \rrbracket \langle \$ \rangle \rho$ 
    subgoal = kerF $\subseteq$ Equal{ $\mathbf{A} = \mathbf{A}$ }{sA} pKq  $\rho$ 
    Goal :  $(\text{free-lift}\{\mathbf{A} = \mathbf{A}\} \rho p) \approx (\text{free-lift}\{\mathbf{A} = \mathbf{A}\} \rho q)$ 
    Goal = trans (sym (fl p)) (trans subgoal (fl q))

homF $\mathfrak{C}$  :  $\text{hom } \mathbb{F}[X] \mathfrak{C}$ 
homF $\mathfrak{C}$  =  $\mid \text{HomFactor } \mathfrak{C} \text{ hom}\mathfrak{C} \text{ homF}[X] \text{ kerF}\subseteq\text{ker}\mathfrak{C} \text{ homF}[X]\text{-is-epic} \mid$ 

ker $\mathfrak{C}\subseteq$ kerF :  $\forall \{p\ q\} \rightarrow (p, q) \in \text{ker} \mid \text{hom}\mathfrak{C} \mid \rightarrow (p, q) \in \text{ker} \mid \text{homF}[X] \mid$ 
ker $\mathfrak{C}\subseteq$ kerF {p}{q} pKq = E $\vdash$ pq
  where
    pqEqual :  $\forall i \rightarrow \text{skEqual } i \{p\}\{q\}$ 
    pqEqual i = goal
      where
        open Environment ( $\mathfrak{A}^+ i$ ) using (  $\llbracket \_ \rrbracket$  )
        open Setoid  $\mathbb{D}[\mathfrak{A}^+ i]$  using (  $\_ \approx \_ ; \text{sym} ; \text{trans}$  )
        goal :  $\llbracket p \rrbracket \langle \$ \rangle \text{snd } \llbracket i \rrbracket \approx \llbracket q \rrbracket \langle \$ \rangle \text{snd } \llbracket i \rrbracket$ 
        goal = trans (free-lift-interp{ $\mathbf{A} = \mid i \mid$ }(snd  $\llbracket i \rrbracket$ ) p)
          (trans (pKq i)(sym (free-lift-interp{ $\mathbf{A} = \mid i \mid$ }(snd  $\llbracket i \rrbracket$ ) q)))
    E $\vdash$ pq :  $\mathcal{K} \vdash X \triangleright p \approx q$ 
    E $\vdash$ pq = AllEqual $\subseteq$ kerF pqEqual

```

```

monF $\mathfrak{C}$  : mon  $\mathbb{F}[X]$   $\mathfrak{C}$ 
monF $\mathfrak{C}$  = | homF $\mathfrak{C}$  | , isMon
where
isMon : IsMon  $\mathbb{F}[X]$   $\mathfrak{C}$  | homF $\mathfrak{C}$  |
isHom isMon = || homF $\mathfrak{C}$  ||
isInjective isMon {p} {q}  $\varphi$ pq = ker $\mathfrak{C}$   $\subseteq$  kerF  $\varphi$ pq

```

Now that we have proved the existence of a monomorphism from  $\mathbb{F}[X]$  to  $\mathfrak{C}$  we can prove that  $\mathbb{F}[X]$  is a subalgebra of  $\mathfrak{C}$ , so belongs to  $S(P\mathcal{K})$ .

```

 $\mathbb{F} \leq \mathfrak{C}$  :  $\mathbb{F}[X] \leq \mathfrak{C}$ 
 $\mathbb{F} \leq \mathfrak{C}$  = mon $\rightarrow \leq$  monF $\mathfrak{C}$ 

SPF :  $\mathbb{F}[X] \in S_{\iota}(P\ell\iota\mathcal{K})$ 
SPF = S-idem SSPF
where
PSC :  $\mathfrak{C} \in P(\alpha \sqcup \rho^a \sqcup \ell) \iota (S\ell\mathcal{K})$ 
PSC =  $\mathcal{I}^+$  , ( $\mathfrak{A}^+$  , (( $\lambda i \rightarrow$  fst || i ||) ,  $\cong$ -refl))
SPC :  $\mathfrak{C} \in S_{\iota}(P\ell\iota\mathcal{K})$ 
SPC = PS $\subseteq$ SP { $\ell = \ell$ } PSC
SSPF :  $\mathbb{F}[X] \in S_{\iota}(S_{\iota}(P\ell\iota\mathcal{K}))$ 
SSPF =  $\mathfrak{C}$  , (SPC ,  $\mathbb{F} \leq \mathfrak{C}$ )

```

## 9 The HSP Theorem

Finally, we are in a position to prove Birkhoff's celebrated variety theorem.

```

module _ { $\mathcal{K}$  : Pred(Algebra  $\alpha \rho^a$ ) ( $\alpha \sqcup \rho^a \sqcup \text{ov } \ell$ )} where
private  $\iota$  = ov( $\alpha \sqcup \rho^a \sqcup \ell$ )
open FreeHom { $\ell = \ell$ }( $\alpha \sqcup \rho^a \sqcup \ell$ ){ $\mathcal{K}$ }
open FreeAlgebra { $\iota = \iota$ }{ $\mathcal{I} = \mathcal{I}$ }  $\mathcal{K}$  using (  $\mathbb{F}[\_]$  )

Birkhoff :  $\forall \mathbf{A} \rightarrow \mathbf{A} \in \text{Mod}(\text{Th}(\mathbf{V}\ell\iota\mathcal{K})) \rightarrow \mathbf{A} \in \mathbf{V}\ell\iota\mathcal{K}$ 
Birkhoff  $\mathbf{A} \text{ ModThA} = \mathbf{V}\cong\text{-lc}\{\alpha\}\{\rho^a\}\{\ell\} \mathcal{K} \mathbf{A} \text{ VIA}$ 
where
open Setoid  $\mathbb{D}[\mathbf{A}]$  using () renaming ( Carrier to  $\mathbf{A}$  )
spFA :  $\mathbb{F}[\mathbf{A}] \in S_{\iota}\iota(P\ell\iota\mathcal{K})$ 
spFA = SPF{ $\ell = \ell$ }  $\mathcal{K}$ 
epiFA : epi  $\mathbb{F}[\mathbf{A}]$  (Lift-Alg  $\mathbf{A} \iota \iota$ )
epiFA =  $\mathbb{F}\text{-ModTh-epi-lift}\{\ell = \ell\} (\lambda \{p\ q\} \rightarrow \text{ModThA}\{p = p\}\{q\})$ 
IAimgFA : Lift-Alg  $\mathbf{A} \iota \iota$  IsHomImageOf  $\mathbb{F}[\mathbf{A}]$ 
IAimgFA = epi $\rightarrow$ ontohom  $\mathbb{F}[\mathbf{A}]$  (Lift-Alg  $\mathbf{A} \iota \iota$ ) epiFA
VIA : Lift-Alg  $\mathbf{A} \iota \iota \in \mathbf{V}\ell\iota\mathcal{K}$ 
VIA =  $\mathbb{F}[\mathbf{A}]$  , spFA , IAimgFA

```

The converse inclusion,  $\mathbf{V}\mathcal{K} \subseteq \text{Mod}(\text{Th}(\mathbf{V}\mathcal{K}))$ , is a simple consequence of the fact that  $\text{Mod Th}$  is a closure operator. Nonetheless, completeness demands that we formalize this inclusion as well, however trivial the proof.

```

module _ { $\mathbf{A}$  : Algebra  $\alpha \rho^a$ } where
Birkhoff-converse :  $\mathbf{A} \in \mathbf{V}\{\alpha\}\{\rho^a\}\{\alpha\}\{\rho^a\}\{\ell\iota\mathcal{K} \rightarrow \mathbf{A} \in \text{Mod}\{X = \mathbb{U}[\mathbf{A}]\}(\text{Th}(\mathbf{V}\ell\iota\mathcal{K}))$ 

```

**Birkhoff-converse**  $\forall A \ pThq = pThq \ A \ \forall A$

We have thus proved that every variety is an equational class.

Readers familiar with the classical formulation of the Birkhoff HSP theorem as an “if and only if” assertion might worry that the proof is still incomplete. However, recall that in the **Varieties.Func.Preservation** module we proved the following identity preservation lemma:

**V-id1** :  $\mathcal{K} \models p \approx \cdot q \rightarrow \forall \mathcal{K} \models p \approx \cdot q$

Thus, if  $\mathcal{K}$  is an equational class—that is, if  $\mathcal{K}$  is the class of algebras satisfying all identities in some set—then  $\forall \mathcal{K} \subseteq \mathcal{K}$ . On the other hand, we proved that  $\forall$  is expansive in the **Varieties.Func.Closure** module:

**V-expa** :  $\mathcal{K} \subseteq \forall \mathcal{K}$

so  $\mathcal{K}$  ( $= \forall \mathcal{K} = \mathbf{HSP} \ \mathcal{K}$ ) is a variety.

Taken together, **V-id1** and **V-expa** constitute formal proof that every equational class is a variety. This completes the formal proof of Birkhoff’s variety theorem.

---

## References

- 1 Clifford Bergman. *Universal Algebra: fundamentals and selected topics*, volume 301 of *Pure and Applied Mathematics (Boca Raton)*. CRC Press, Boca Raton, FL, 2012.
- 2 Venanzio Capretta. Universal algebra in type theory. In *Theorem proving in higher order logics (Nice, 1999)*, volume 1690 of *Lecture Notes in Comput. Sci.*, pages 131–148. Springer, Berlin, 1999. doi:10.1007/3-540-48256-3\_10.
- 3 Emmanuel Gunther, Alejandro Gadea, and Miguel Pagano. Formalization of universal algebra in Agda. *Electronic Notes in Theoretical Computer Science*, 338:147 – 166, 2018. The 12th Workshop on Logical and Semantic Frameworks, with Applications (LSFA 2017). doi:https://doi.org/10.1016/j.entcs.2018.10.010.
- 4 Bas Spitters and Eelis Van der Weegen. Type classes for mathematics in type theory. *CoRR*, abs/1102.1323, 2011. arXiv:1102.1323.